

# Lógica de primer orden en Haskell

Eduardo Paluzo

---

Grupo de Lógica Computacional

Dpto. de Ciencias de la Computación e Inteligencia Artificial

Universidad de Sevilla

Sevilla, 16 de junio de 2016 (Versión de 2 de abril de 2017)

Esta obra está bajo una licencia Reconocimiento–NoComercial–CompartirIgual 2.5 Spain de Creative Commons.

**Se permite:**

- copiar, distribuir y comunicar públicamente la obra
- hacer obras derivadas

**Bajo las condiciones siguientes:**



**Reconocimiento.** Debe reconocer los créditos de la obra de la manera especificada por el autor.



**No comercial.** No puede utilizar esta obra para fines comerciales.



**Compartir bajo la misma licencia.** Si altera o transforma esta obra, o genera una obra derivada, sólo puede distribuir la obra generada bajo una licencia idéntica a ésta.

- Al reutilizar o distribuir la obra, tiene que dejar bien claro los términos de la licencia de esta obra.
- alguna de estas condiciones puede no aplicarse si se obtiene el permiso del titular de los derechos de autor.

Esto es un resumen del texto legal (la licencia completa). Para ver una copia de esta licencia, visite <http://creativecommons.org/licenses/by-nc-sa/2.5/es/> o envíe una carta a Creative Commons, 559 Nathan Abbott Way, Stanford, California 94305, USA.

# Índice general

<b>Introducción</b>	<b>7</b>
<b>I Semántica computacional y teoría de tipos</b>	<b>9</b>
<b>1 Programación funcional con Haskell</b>	<b>11</b>
1.1 Introducción a Haskell . . . . .	11
1.1.1 Comprensión de listas . . . . .	13
1.1.2 Funciones map y filter . . . . .	13
1.1.3 n-uplas . . . . .	14
1.1.4 Conjunción, disyunción y cuantificación . . . . .	15
1.1.5 Plegados especiales foldr y foldl . . . . .	16
1.1.6 Teoría de tipos . . . . .	18
1.1.7 Generador de tipos en Haskell: Descripción de funciones . . . . .	19
1.2 Librería Data.Map . . . . .	19
<b>2 Sintaxis y semántica de la lógica de primer orden</b>	<b>21</b>
2.1 Representación de modelos . . . . .	21
2.2 Lógica de primer orden en Haskell . . . . .	24
2.3 Evaluación de fórmulas . . . . .	27
2.4 Términos funcionales . . . . .	29
2.4.1 Generadores . . . . .	35
2.4.2 Otros conceptos de la lógica de primer orden . . . . .	38
<b>3 Deducción natural</b>	<b>41</b>
3.1 Sustitución . . . . .	41
3.2 Sustitución mediante diccionarios . . . . .	46
3.3 Reglas de deducción natural . . . . .	46
3.3.1 Reglas de la conjunción . . . . .	48
3.3.2 Reglas de eliminación del condicional . . . . .	49
3.3.3 Reglas de la disyunción . . . . .	50

3.3.4	Reglas de la negación . . . . .	50
3.3.5	Reglas del bicondicional . . . . .	51
3.3.6	Regla derivada de Modus Tollens(MT) . . . . .	51
3.3.7	Reglas de la doble negación . . . . .	52
3.3.8	Regla de Reducción al absurdo . . . . .	52
3.3.9	Ley del tercio excluido . . . . .	53
3.3.10	Reglas del cuantificador universal . . . . .	53
3.3.11	Reglas del cuantificador existencial . . . . .	53
3.3.12	Ejemplos . . . . .	54
<b>4</b>	<b>Prueba de teoremas en lógica de predicados</b>	<b>55</b>
4.1	Unificación . . . . .	55
4.2	Skolem . . . . .	56
4.2.1	Formas normales . . . . .	56
4.2.2	Forma rectificada . . . . .	61
4.2.3	Forma normal prenexa . . . . .	62
4.2.4	Forma normal prenexa conjuntiva . . . . .	63
4.2.5	Forma de Skolem . . . . .	64
4.3	Forma clausal . . . . .	66
4.4	Tableros semánticos . . . . .	68
<b>5</b>	<b>Modelos de Herbrand</b>	<b>79</b>
5.1	Universo de Herbrand . . . . .	79
5.2	Base de Herbrand . . . . .	83
5.3	Interpretaciones de Herbrand . . . . .	86
5.4	Modelos de Herbrand . . . . .	87
5.5	Consistencia mediante modelos de Herbrand . . . . .	88
5.6	Extensiones de Herbrand . . . . .	89
<b>6</b>	<b>Modelos de Herbrand alternativo</b>	<b>91</b>
6.1	Universo de Herbrand . . . . .	91
6.2	Base de Herbrand . . . . .	97
6.3	Interpretacion de Herbrand . . . . .	99
6.4	Modelos de Herbrand . . . . .	100
<b>7</b>	<b>Resolución en lógica de primer orden</b>	<b>103</b>
7.1	Ampliación de la forma clausal . . . . .	103
<b>8</b>	<b>Correspondencia de Curry-Howard</b>	<b>107</b>

---

<b>II</b>	<b>Sistemas utilizados</b>	<b>111</b>
<b>9</b>	<b>Apéndice: GitHub</b>	<b>113</b>
9.1	Crear una cuenta . . . . .	113
9.2	Crear un repositorio . . . . .	113
9.3	Conexión . . . . .	113
9.4	Pull y push . . . . .	114
9.5	Ramificaciones (“branches”) . . . . .	115
	<b>Bibliografía</b>	<b>116</b>
	<b>Índice de definiciones</b>	<b>116</b>
	<b>Lista de tareas pendientes</b>	<b>118</b>



# Introducción

El objetivo del trabajo es la implementación de los algoritmos de la lógica de primer orden en Haskell. Consta de dos partes:

- La primera parte consiste en la implementación en Haskell de la teoría impartida en la asignatura [Lógica matemática y fundamentos](#)<sup>1</sup> ([?]). Para ello, se lleva a cabo la adaptación de los programas del libro de J. van Eijck [Computational semantics and type theory](#)<sup>2</sup> ([?]) y su correspondiente teoría.
- En la segunda parte se comentan aquellos sistemas empleados en la elaboración del trabajo.

---

<sup>1</sup><https://www.cs.us.es/~jalonso/cursos/lmf-15>

<sup>2</sup><http://citeseerx.ist.psu.edu/viewdoc/download?doi=10.1.1.467.1441&rep=rep1&type=pdf>





## **Parte I**

# **Semántica computacional y teoría de tipos**



# Capítulo 1

## Programación funcional con Haskell

En este capítulo se hace una breve introducción a la programación funcional en Haskell suficiente para entender su aplicación en los siguientes capítulos. Para una introducción más amplia se pueden consultar los apuntes de la asignatura de Informática de 1º del Grado en Matemáticas ([?]). También se puede emplear como lectura complementaria, y se ha empleado para algunas definiciones del trabajo ([?])

El contenido de este capítulo se encuentra en el módulo PFH

```
module PFH where
import Data.List
```

### 1.1. Introducción a Haskell

En esta sección, se introducirán funciones básicas para la programación en Haskell. Como método didáctico, se empleará la definición de funciones ejemplos, así como la redefinición de funciones que Haskell ya tiene predefinidas, con el objetivo de que el lector aprenda *“a montar en bici, montando”*.

A continuación se muestra la definición del que constituye nuestro primer ejemplo. (cuadrado x) es el cuadrado de x. Por ejemplo,

```
ghci> cuadrado 3
9
ghci> cuadrado 4
16
```

La definición es

```
cuadrado :: Int -> Int
cuadrado x = x * x
```

Definimos otra función en Haskell. (raizCuadrada x) es la raíz cuadrada entera de x. Por ejemplo,

```
ghci> raizCuadrada 9
3
ghci> raizCuadrada 8
2
```

La definición es

```
raizCuadrada :: Int -> Int
raizCuadrada x = last [y | y <- [1..x], y*y <= x]
```

Posteriormente, definimos funciones que determinen si un elemento  $x$  cumple una cierta propiedad. Este es el caso de la propiedad 'ser divisible por  $n$ ', donde  $n$  será un número cualquiera.

```
ghci> 15 'divisiblePor' 5
True
```

La definición es

```
divisiblePor :: Int -> Int -> Bool
divisiblePor x n = x 'rem' n == 0
```

Hasta ahora hemos trabajado con los tipos de datos `Int` y `Bool`; es decir, números enteros y booleanos respectivamente. Pero también se puede trabajar con otros tipos de datos como son las cadenas de caracteres, que son tipo `[Char]` o `String`. Definimos la función (`contieneLaLetra xs l`) que identifica si una palabra contiene una cierta letra  $l$  dada. Por ejemplo,

```
ghci> "descartes" 'contieneLaLetra' 'e'
True
ghci> "topologia" 'contieneLaLetra' 'm'
False
```

Y su definición es

```
contieneLaLetra :: String -> Char -> Bool
contieneLaLetra [] _ = False
contieneLaLetra (x:xs) l = x == l || contieneLaLetra xs l
```

### 1.1.1. Comprensión de listas

Las listas son una representación de un conjunto ordenado de elementos. Dichos elementos pueden ser de cualquier tipo, ya sean `Int`, `Bool`, `Char`, ... Siempre y cuando todos los elementos de dicha lista compartan tipo. En Haskell las listas se representan

```
ghci> [1,2,3,4]
[1,2,3,4]
ghci> [1..4]
[1,2,3,4]
```

Una lista por comprensión es parecido a su expresión como conjunto:

$$\{x | x \in A, P(x)\}$$

Se puede leer de manera intuitiva como: “tomar aquellos  $x$  del conjunto  $A$  tales que cumplen una cierta propiedad  $P$ ”. En Haskell se representa

$$[x | x \leftarrow \text{lista}, \text{condiciones que debe cumplir}]$$

Algunos ejemplos son:

```
ghci> [n | n <- [0 .. 10], even n]
[0,2,4,6,8,10]
ghci> [x | x <- ["descartes","pitagoras","gauss"], x 'contieneLaLetra' 'e']
["descartes"]
```

*Nota 1.1.1.* En los distintos ejemplos hemos visto que se pueden componer funciones ya definidas.

Otro ejemplo, de una mayor dificultad, es la construcción de variaciones con repeticiones de una lista. Se define (`variacionesR n xs`)

```
variacionesR :: Int -> [a] -> [[a]]
variacionesR _ [] = [[]]
variacionesR 0 _ = [[]]
variacionesR k xs =
    [z:ys | z <- xs, ys <- variacionesR (k-1) xs]
```

*Nota 1.1.2.* La función `variacionesR` será útil en capítulos posteriores.

### 1.1.2. Funciones `map` y `filter`

Introducimos un par de funciones de mucha relevancia en el uso de listas en Haskell. Son funciones que se denominan de orden superior.

La función (`map f xs`) aplica una función `f` a cada uno de los elementos de la lista `xs`. Por ejemplo,

```
ghci> map ('divisiblePor' 4) [8,12,3,9,16]
[True,True,False,False,True]
ghci> map ('div' 4) [8,12,3,9,16]
[2,3,0,2,4]
ghci> map ('div' 4) [x | x <- [8,12,3,9,16], x 'divisiblePor' 4]
[2,3,4]
```

Dicha función está predefinida en el paquete `Data.List`, nosotros daremos una definición denotándola con el nombre `(aplicafun f xs)`, y su definición es

```
aplicafun :: (a -> b) -> [a] -> [b]
aplicafun f []      = []
aplicafun f (x:xs) = f x : aplicafun f xs
```

La función `(filter p xs)` es la lista de los elementos de `xs` que cumplen la propiedad `p`. Por ejemplo,

```
ghci> filter (<5) [1,5,7,2,3]
[1,2,3]
```

La función `filter` al igual que la función `map` está definida en el paquete `Data.List`, pero nosotros la denotaremos como `(aquellosQuecumplen p xs)`. Y su definición es

```
aquellosQuecumplen :: (a -> Bool) -> [a] -> [a]
aquellosQuecumplen p [] = []
aquellosQuecumplen p (x:xs) | p x      = x : aquellosQuecumplen p xs
                             | otherwise = aquellosQuecumplen p xs
```

En esta última definición hemos introducido las ecuaciones por guardas, representadas por `|`. Otro ejemplo más simple del uso de guardas es el siguiente

$$g(x) = \begin{cases} 5, & \text{si } x \neq 0 \\ 0, & \text{en caso contrario} \end{cases}$$

Que en Haskell sería

```
g :: Int -> Int
g x | x /= 0    = 5
    | otherwise = 0
```

### 1.1.3. n-uplas

Una  $n$ -upla es un elemento del tipo  $(a_1, \dots, a_n)$  y existen una serie de funciones para el empleo de los pares  $(a_1, a_2)$ . Dichas funciones están predefinidas bajo los nombres `fst` y `snd`, y las redefinimos como `(primerElemento)` y `(segundoElemento)` respectivamente.

```
primerElemento :: (a,b) -> a
primerElemento (x,_) = x

segundoElemento :: (a,b) -> b
segundoElemento (_,y) = y
```

### 1.1.4. Conjunción, disyunción y cuantificación

En Haskell, la conjunción está definida mediante el operador `&&`, y se puede generalizar a listas mediante la función predefinida (`and xs`) que nosotros redefinimos denotándola (`conjuncion xs`). Su definición es

```
conjuncion :: [Bool] -> Bool
conjuncion []      = True
conjuncion (x:xs) = x && conjuncion xs
```

Dicha función es aplicada a una lista de booleanos y ejecuta una conjunción generalizada.

La disyunción en Haskell se representa mediante el operador `||`, y se generaliza a listas mediante una función predefinida (`or xs`) que nosotros redefinimos con el nombre (`disyuncion xs`). Su definición es

```
disyuncion :: [Bool] -> Bool
disyuncion []      = False
disyuncion (x:xs) = x || disyuncion xs
```

Posteriormente, basándonos en estas generalizaciones de operadores lógicos se definen los siguientes cuantificadores, que están predefinidos como (`any p xs`) y (`all p xs`) en Haskell, y que nosotros redefinimos bajo los nombres (`algun p xs`) y (`todos p xs`). Se definen

```
algun, todos :: (a -> Bool) -> [a] -> Bool
algun p = disyuncion . aplicafun p
todos p = conjuncion . aplicafun p
```

*Nota 1.1.3.* Hemos empleando composición de funciones para la definición de (`algun`) y (`todos`). Se representa mediante `.`, y se omite el argumento de entrada común a todas las funciones.

En matemáticas, estas funciones representan los cuantificadores lógicos  $\exists$  y  $\forall$ , y determinan si alguno de los elementos de una lista cumple una cierta propiedad, y si

todos los elementos cumplen una determinada propiedad respectivamente. Por ejemplo.

$\forall x \in \{0, \dots, 10\}$  se cumple que  $x < 7$ . Es Falso

En Haskell se aplicaría la función (`todos p xs`) de la siguiente forma

```
ghci> todos (<7) [0..10]
False
```

Finalmente, definimos las funciones (`pertenece x xs`) y (`noPertenece x xs`)

```
pertenece, noPertenece :: Eq a => a -> [a] -> Bool
pertenece    = algun . (==)
noPertenece  = todos . (/=)
```

Estas funciones determinan si un elemento `x` pertenece a una lista `xs` o no.

### 1.1.5. Plegados especiales `foldr` y `foldl`

No nos hemos centrado en una explicación de la recursión pero la hemos empleado de forma intuitiva. En el caso de la recursión sobre listas, hay que distinguir un caso base; es decir, asegurarnos de que tiene fin. Un ejemplo de recursión es la función (`factorial x`), que definimos

```
factorial :: Int -> Int
factorial 0 = 1
factorial x = x*(x-1)
```

Añadimos una función recursiva sobre listas, como puede ser (`sumaLaLista xs`)

```
sumaLaLista :: Num a => [a] -> a
sumaLaLista []      = 0
sumaLaLista (x:xs) = x + sumaLaLista xs
```

Tras este preámbulo sobre recursión, introducimos la función (`foldr f z xs`) que no es más que una recursión sobre listas o plegado por la derecha, la definimos bajo el nombre (`plegadoPorlaDerecha f z xs`)

```
plegadoPorlaDerecha :: (a -> b -> b) -> b -> [a] -> b
plegadoPorlaDerecha f z []      = z
plegadoPorlaDerecha f z (x:xs) = f x (plegadoPorlaDerecha f z xs)
```

Un ejemplo de aplicación es el producto de los elementos o la suma de los elementos de una lista



```
ghci> plegadoPorlaDerecha (*) 1 [1,2,3]
6
ghci> plegadoPorlaDerecha (+) 0 [1,2,3]
6
```

Un esquema informal del funcionamiento de `plegadoPorlaDerecha` es

$$\text{plegadoPorlaDerecha } (\otimes) z [x_1, x_2, \dots, x_n] := x_1 \otimes (x_2 \otimes (\dots (x_n \otimes z) \dots))$$

*Nota 1.1.4.*  $\otimes$  representa una operación cualquiera.

Por lo tanto, podemos dar otras definiciones para las funciones `(conjuncion xs)` y `(disyuncion xs)`

```
conjuncion1, disyuncion1 :: [Bool] -> Bool
conjuncion1 = plegadoPorlaDerecha (&&) True
disyuncion1 = plegadoPorlaDerecha (||) False
```

Hemos definido `plegadoPorlaDerecha`, ahora el lector ya intuirá que `(foldl f z xs)` no es más que una función que pliega por la izquierda. Definimos `(plegadoPorlaIzquierda f z xs)`

```
plegadoPorlaIzquierda :: (a -> b -> a) -> a -> [b] -> a
plegadoPorlaIzquierda f z [] = z
plegadoPorlaIzquierda f z (x:xs) = plegadoPorlaIzquierda f (f z x) xs
```

De manera análoga a `foldr` mostramos un esquema informal para facilitar la comprensión

$$\text{plegadoPorlaIzquierda } (\otimes) z [x_1, x_2, \dots, x_n] := (\dots ((z \otimes x_1) \otimes x_2) \otimes \dots) \otimes x_n$$

Definamos una función ejemplo como es la inversa de una lista. Está predefinida bajo el nombre `(reverse xs)` y nosotros la redefinimos como `(listaInversa xs)`

```
listaInversa :: [a] -> [a]
listaInversa = plegadoPorlaIzquierda (\xs x -> x:xs) []
```

Por ejemplo

```
ghci> listaInversa [1,2,3,4,5]
[5,4,3,2,1]
```

Podríamos comprobar por ejemplo si la frase 'Yo dono rosas, oro no doy' es un palíndromo

```
ghci> listaInversa "yodonorosasonodoy"
"yodonorosasonodoy"
ghci> listaInversa "yodonorosasonodoy" == "yodonorosasonodoy"
True
```

### 1.1.6. Teoría de tipos

#### Notación $\lambda$

Cuando hablamos de notación lambda simplemente nos referimos a expresiones del tipo  $\lambda x. x+2$ . La notación viene del  $\lambda$  *Calculus* y se escribiría  $\lambda x. x+2$ . Los diseñadores de Haskell tomaron el símbolo  $\lambda$  debido a su parecido con  $\lambda$  y por ser fácil y rápido de teclear. Una función ejemplo es `(divideEntre2 xs)`

```
divideEntre2 :: Fractional b => [b] -> [b]
divideEntre2 xs = map (\x -> x/2) xs
```

Para una información más amplia recomiendo consultar ([?])

#### Representación de un dominio de entidades

**Definición 1.1.1.** Un **dominio de entidades** es un conjunto de individuos cuyas propiedades son objeto de estudio para una clasificación

Construimos un ejemplo de un dominio de entidades compuesto por las letras del abecedario, declarando el tipo de dato `Entidades` contenido en el módulo `Dominio`

```
module Dominio where

data Entidades = A | B | C | D | E | F | G
               | H | I | J | K | L | M | N
               | O | P | Q | R | S | T | U
               | V | W | X | Y | Z | Inespecifico
               deriving (Eq, Bounded, Enum)
```

Se añade `deriving (Eq, Bounded, Enum)` para establecer relaciones de igualdad entre las entidades (`Eq`), una acotación (`Bounded`) y enumeración de los elementos (`Enum`).

Para mostrarlas por pantalla, definimos las entidades en la clase (`Show`) de la siguiente forma

```
instance Show Entidades where
    show A = "A"; show B = "B"; show C = "C";
    show D = "D"; show E = "E"; show F = "F";
    show G = "G"; show H = "H"; show I = "I";
    show J = "J"; show K = "K"; show L = "L";
    show M = "M"; show N = "N"; show O = "O";
    show P = "P"; show Q = "Q"; show R = "R";
    show S = "S"; show T = "T"; show U = "U";
    show V = "V"; show W = "W"; show X = "X";
    show Y = "Y"; show Z = "Z"; show Inespecifico = "*"
```

Colocamos todas las entidades en una lista

```
entidades :: [Entidades]
entidades = [minBound..maxBound]
```

De manera que si lo ejecutamos

```
ghci> entidades
[A,B,C,D,E,F,G,H,I,J,K,L,M,N,O,P,Q,R,S,T,U,V,W,X,Y,Z,*]
```

### 1.1.7. Generador de tipos en Haskell: Descripción de funciones

En esta sección se introducirán y describirán funciones útiles en la generación de ejemplos en tipos de datos abstractos. Estos generadores son útiles para las comprobación de propiedades con QuickCheck

## 1.2. Librería Data.Map

Introducimos la librería Data.Map cuya función es el trabajo con diccionarios, permitiendo tanto la construcción de estos diccionarios, como su modificación y acceso a la información.

```
module Map where
import Data.List
import Data.Map (Map)
import qualified Data.Map as M
import Text.PrettyPrint
import Text.PrettyPrint.GenericPretty
```

Debido a que mucha de sus funciones tienen nombres coincidentes con algunas ya definidas en Prelude, es necesario importarla renombrándola, de la siguiente manera: `import qualified Data.Map as M`. Eso implica que cuando llamemos a una función de esta librería, tendremos que hacerlo poniendo M. (función).

Los diccionarios son del tipo `M k` y la forma de construirlos es mediante la función `(M.fromList)` seguida de una lista de pares.

```
-- | Ejemplos:
-- >>> :type M.fromList
-- M.fromList :: Ord k => [(k, a)] -> Map k a
--
-- >>> M.fromList [(1,"Pablo"),(10,"Elisabeth"),(7,"Cristina"),(0,"Luis")]
-- fromList [(0,"Luis"),(1,"Pablo"),(7,"Cristina"),(10,"Elisabeth")]
```

Una vez creada un diccionario, podemos acceder a la información registrada en él, y modificarla.

El operador (`M.!`) sirve para acceder a elementos del diccionario.

```
-- | Ejemplos:
-- >>> let l = M.fromList [(1,"Pablo"),(10,"Elisabeth"),(7,"Cristina"),(0,"Luis")]
-- >>> l M.! 1
-- "Pablo"
-- >>> l M.! 10
-- "Elisabeth"
```

La función (`M.size`) devuelve el tamaño del diccionario; es decir, su número de elementos.

```
-- | Ejemplos
-- >>> let l = M.fromList [(1,"Pablo"),(10,"Elisabeth"),(7,"Cristina"),(0,"Luis")]
-- >>> M.size l
-- 4
```

La función (`M.insert`) registra un elemento en el diccionario.

```
-- | Ejemplos
-- >>> let l = M.fromList [(1,"Pablo"),(10,"Elisabeth"),(7,"Cristina"),(0,"Luis")]
-- >>> M.insert 8 "Jesus" l
-- fromList [(0,"Luis"),(1,"Pablo"),(7,"Cristina"),(8,"Jesus"),(10,"Elisabeth")]
```

sección en proceso

# Capítulo 2

## Sintaxis y semántica de la lógica de primer orden

El lenguaje de la lógica de primer orden está compuesto por

1. Variables proposicionales  $p, q, r, \dots$
2. Conectivas lógicas:

$\neg$	Negación
$\vee$	Disyunción
$\wedge$	Conjunción
$\rightarrow$	Condicional
$\leftrightarrow$	Bicondicional

3. Símbolos auxiliares " $(, )$ "
4. Cuantificadores:  $\forall$  (Universal) y  $\exists$  (Existencial)
5. Símbolo de igualdad:  $=$
6. Constantes:  $a, b, \dots, a_1, a_2, \dots$
7. Símbolos de relación:  $P, Q, R, \dots$
8. Símbolos de función:  $f, g, h, \dots$

### 2.1. Representación de modelos

El contenido de esta sección se encuentra en el módulo `Modelo`.

```

module Modelo where
import Dominio
import PFH

```

La lógica de primer orden permite dar una representación al conocimiento. Nosotros trabajaremos con modelos a través de un dominio de entidades; en concreto, aquellas del módulo `Dominio`. Cada entidad de dicho módulo representa un sujeto. Cada sujeto tendrá distintas propiedades.

En secciones posteriores se definirá un modelo lógico. Aquí empleamos el término modelo como una modelización o representación de la realidad.

Damos un ejemplo de predicados lógicos para la clasificación botánica. La cual no es completa, pero nos da una idea de la manera de una representación lógica.

Primero definimos los elementos que pretendemos clasificar, y que cumplirán los predicados. Con este fin, definimos una función para cada elemento del dominio de entidades.

```

adelfas, aloeVera, boletus, cedro, chlorella, girasol, guisante, helecho,
  hepatica, jaramago, jazmin, lenteja, loto, magnolia, maiz, margarita,
  musgo, olivo, pino, pita, posidonia, rosa, sargazo, scenedesmus,
  tomate, trigo
  :: Entidades
adelfas      = U
aloeVera     = L
boletus      = W
cedro        = A
chlorella    = Z
girasol      = Y
guisante     = S
helecho      = E
hepatica     = V
jaramago     = X
jazmin       = Q
lenteja      = R
loto         = T
magnolia     = O
maiz         = F
margarita    = K
musgo        = D
olivo        = C
pino         = J
pita         = M
posidonia    = H
rosa         = P
sargazo      = I
scenedesmus  = B

```

```
tomate      = N
trigo       = G
```

Una vez que ya tenemos todos los elementos a clasificar definidos, se procede a la interpretación de los predicados. Es decir, una clasificación de aquellos elementos que cumplen un cierto predicado.

**Definición 2.1.1.** Un **predicado** es una oración narrativa que puede ser verdadera o falsa.

```
acuatica, algasVerdes, angiosperma, asterida, briofita, cromista,
  crucifera, dicotiledonea, gimnosperma, hongo, leguminosa,
  monoaperturada, monocotiledonea, rosida, terrestre,
  triaperturada, unicelular
:: Entidades -> Bool
acuatica      = ('pertenece' [B,H,I,T,Z])
algasVerdes    = ('pertenece' [B,Z])
angiosperma    = ('pertenece' [C,F,G,H,K,L,M,N,O,P,Q,R,S,T,U,X,Y])
asterida       = ('pertenece' [C,K,N,Q,U,Y])
briofita       = ('pertenece' [D,V])
cromista       = ('pertenece' [I])
crucifera      = ('pertenece' [X])
dicotiledonea  = ('pertenece' [C,K,N,O,P,Q,R,S,T,U,X,Y])
gimnosperma    = ('pertenece' [A,J])
hongo          = ('pertenece' [W])
leguminosa     = ('pertenece' [R,S])
monoaperturada = ('pertenece' [F,G,H,L,M,O])
monocotiledonea = ('pertenece' [F,G,H,L,M])
rosida         = ('pertenece' [P])
terrestre      =
  ('pertenece' [A,C,D,E,F,G,J,K,L,M,N,O,P,Q,R,S,U,V,W,X,Y])
triaperturada  = ('pertenece' [C,K,N,P,Q,R,S,T,U,X,Y])
unicelular     = ('pertenece' [B,Z])
```

Por ejemplo, podríamos comprobar si el `scenedesmus` es `gimnosperma`

```
ghci> gimnosperma scenedesmus
False
```

Esto nos puede facilitar establecer una jerarquía en la clasificación, por ejemplo (espermatofitas); es decir, plantas con semillas.

```
espermatofitas :: Entidades -> Bool
espermatofitas x = angiosperma x || gimnosperma x
```

## 2.2. Lógica de primer orden en Haskell

El contenido de esta sección se encuentra en el módulo LPH. Se pretende asentar las bases de la lógica de primer orden y su implementación en Haskell, con el objetivo de construir los cimientos para las posteriores implementaciones de algoritmos en los siguientes capítulos.

```
{-# LANGUAGE DeriveGeneric #-}

module LPH where

import Dominio
import Modelo
import Data.List
import Test.QuickCheck
import Text.PrettyPrint
import Text.PrettyPrint.GenericPretty
```

Los elementos básicos de las fórmulas en la lógica de primer orden, así como en la lógica proposicional son las variables. Por ello, definimos un tipo de dato para las variables.

Una variable estará compuesta por un nombre y un índice, es decir, nombraremos las variables como  $x_1, a_1, \dots$

El tipo de dato para el nombre lo definimos como una lista de caracteres

```
type Nombre = String
```

El tipo de dato para los índices lo definimos como lista de enteros.

```
type Indice = [Int]
```

Quedando el tipo de dato compuesto Variable como

```
data Variable = Variable Nombre Indice
  deriving (Eq,Ord,Generic)
```

Para una visualización agradable en pantalla se define su representación en la clase Show.

```
instance Show Variable where
  show (Variable nombre []) = nombre
  show (Variable nombre [i]) = nombre ++ show i
  show (Variable nombre is) = nombre ++ showInts is
    where showInts [] = ""
```



```
        showInts [i]      = show i
        showInts (i:is') = show i ++ "_" ++ showInts is'

instance Out Variable where
    doc      = text . show
    docPrec _ = doc
```

Mostramos algunos ejemplos de definición de variables

```
x, y, z :: Variable
x = Variable "x" []
y = Variable "y" []
z = Variable "z" []
u = Variable "u" []
```

Y definimos también variables empleando índices

```
a1, a2, a3 :: Variable
a1 = Variable "a" [1]
a2 = Variable "a" [2]
a3 = Variable "a" [3]
```

De manera que su visualización será

```
ghci> x
x
ghci> y
y
ghci> a1
a1
ghci> a2
a2
```

**Definición 2.2.1.** Se dice que  $F$  es una **fórmula** si satisface la siguiente definición inductiva

1. Las variables proposicionales son fórmulas atómicas.
2. Si  $F$  y  $G$  son fórmulas, entonces  $\neg F$ ,  $(F \wedge G)$ ,  $(F \vee G)$ ,  $(F \rightarrow G)$  y  $(F \leftrightarrow G)$  son fórmulas.

Se define un tipo de dato para las fórmulas lógicas de primer orden.

```

data Formula = Atomo Nombre [Variable]
              | Igual Variable Variable
              | Negacion Formula
              | Implica Formula Formula
              | Equivalente Formula Formula
              | Conjuncion [Formula]
              | Disyuncion [Formula]
              | ParaTodo Variable Formula
              | Existe Variable Formula
  deriving (Eq,Ord)

```

Y se define una visualización en la clase Show

```

instance Show Formula where
  show (Atomo r [])      = r
  show (Atomo r vs)      = r ++ show vs
  show (Igual t1 t2)     = show t1 ++ "≡" ++ show t2
  show (Negacion formula) = '¬' : show formula
  show (Implica f1 f2)   = "(" ++ show f1 ++ "⇒" ++ show f2 ++ ")"
  show (Equivalente f1 f2) = "(" ++ show f1 ++ "⇔" ++ show f2 ++ ")"
  show (Conjuncion [])   = "true"
  show (Conjuncion (f:fs)) = "(" ++ show f ++ "∧" ++ show fs ++ ")"
  show (Disyuncion [])   = "false"
  show (Disyuncion (f:fs)) = "(" ++ show f ++ "∨" ++ show fs ++ ")"
  show (ParaTodo v f)    = "∀" ++ show v ++ (' ': show f)
  show (Existe v f)      = "∃" ++ show v ++ (' ': show f)

```

Como ejemplo podemos representar las propiedades reflexiva y simétrica.

```

-- | Ejemplos
-- >>> reflexiva
-- ∀x R[x,x]
-- >>> simetrica
-- ∀x ∀y (R[x,y]⇒R[y,x])
reflexiva, simetrica :: Formula
reflexiva = ParaTodo x (Atomo "R" [x,x])
simetrica = ParaTodo x (ParaTodo y ( Atomo "R" [x,y] 'Implica'
                                     Atomo "R" [y,x]))

```

**Definición 2.2.2.** Una **estructura del lenguaje**  $L$  es un par  $\mathcal{I} = (\mathcal{U}, I)$  tal que

1.  $\mathcal{U}$  es un conjunto no vacío, denominado universo.
2.  $I$  es una función con dominio el conjunto de símbolos propios de  $L$ .  $L : \text{Símbolos} \rightarrow \text{Símbolos}$  tal que

- si  $c$  es una constante de  $L$ , entonces  $I(c) \in \mathcal{U}$
- si  $f$  es un símbolo de función  $n$ -aria de  $L$ , entonces  $I(f) : \mathcal{U}^n \rightarrow \mathcal{U}$
- si  $P$  es un símbolo de relación 0-aria de  $L$ , entonces  $I(P) \in \{1, \}$
- si  $R$  es un símbolo de relación  $n$ -aria de  $L$ , entonces  $I(R) \subseteq \mathcal{U}^n$

Para el manejo de estructuras del lenguaje, vamos a definir tipos de datos para cada uno de sus elementos.

Definimos el tipo de dato relativo al universo como una lista de elementos.

```
type Universo a = [a]
```

**Definición 2.2.3.** Una **asignación** es una función que hace corresponder a cada variable un elemento del universo.

Se define un tipo de dato para las asignaciones

```
type Asignacion a = Variable -> a
```

Necesitamos definir una asignación para los ejemplos. Tomamos una asignación constante muy sencilla.

```
asignacion :: a -> Entidades
asignacion v = A
```

## 2.3. Evaluación de fórmulas

En esta sección se pretende interpretar fórmulas. Una interpretación toma valores para las variables proposicionales, y se evalúan en una fórmula, determinando si la fórmula es verdadera o falsa, bajo esa interpretación.

**Definición 2.3.1.** Una **interpretación proposicional** es una aplicación  $I : VP \rightarrow Bool$ , donde  $VP$  representa el conjunto de las variables proposicionales.

A continuación, presentamos una tabla de valores de las distintas conectivas lógicas según las interpretaciones de  $P$  y  $Q$ .  $P$  y  $Q$  tienen dos posibles interpretaciones: Falso o verdadero. Falso lo representamos mediante el 0, y verdadero mediante el 1.

P	Q	$P \wedge Q$	$P \vee Q$	$P \rightarrow Q$	$P \leftrightarrow Q$
1	1	1	1	1	1
1	0	0	1	0	0
0	1	0	1	1	0
0	0	0	0	1	1

El valor de  $(\text{sustituye } s \ x \ d)$  es la asignación obtenida a partir de la asignación  $s$  pero con  $x$  interpretado como  $d$ ,

$$\text{sustituye } (s(t), x, d, v) = \begin{cases} d, & \text{si } x = v \\ s(v), & \text{en caso contrario} \end{cases}$$

```
-- | Ejemplos
-- >>> sustituye asignacion y B z
-- A
-- >>> sustituye asignacion y B y
-- B
sustituye :: Asignacion a -> Variable -> a -> Asignacion a
sustituye s x d v | x == v      = d
                  | otherwise = s v
```

**Definición 2.3.2.** Una **interpretación de una estructura del lenguaje** es un par  $(\mathcal{I}, A)$  formado por una estructura del lenguaje y una asignación  $A$ .

Definimos un tipo de dato para las interpretaciones de los símbolos de relación.

```
type InterpretacionR a = String -> [a] -> Bool
```

Definimos la función  $(\text{valor } u \ i \ s \ \text{form})$  que calcula el valor de una fórmula en un universo  $u$ , con una interpretación  $i$ , respecto de la asignación  $s$ .

```
valor :: Eq a =>
  Universo a -> InterpretacionR a -> Asignacion a
  -> Formula -> Bool
valor _ i s (Atomo r vs)      = i r (map s vs)
valor _ _ s (Igual v1 v2)     = s v1 == s v2
valor u i s (Negacion f)      = not (valor u i s f)
valor u i s (Implica f1 f2)   = valor u i s f1 <= valor u i s f2
valor u i s (Equivalente f1 f2) = valor u i s f1 == valor u i s f2
valor u i s (Conjuncion fs)    = all (valor u i s) fs
valor u i s (Disyuncion fs)    = any (valor u i s) fs
valor u i s (ParaTodo v f)    = and [valor u i (sustituye s v d) f
                                     | d <- u]
valor u i s (Existe v f)      = or  [valor u i (sustituye s v d) f
                                     | d <- u]
```

Empleando las entidades y los predicados definidos en los módulos Dominio y Modelo, establecemos un ejemplo del valor de una interpretación en una fórmula.

Primero definimos la fórmula a interpretar, `formula1`, y dos posibles interpretaciones `interpretacion1` e `interpretacion2`.

```

formula1 :: Formula
formula1 = ParaTodo x (Disyuncion [Atomo "P" [x],Atomo "Q" [x]])

interpretacion1 :: String -> [Entidades] -> Bool
interpretacion1 "P" [x] = angiosperma x
interpretacion1 "Q" [x] = gimnosperma x
interpretacion1 _ _      = False

interpretacion2 :: String -> [Entidades] -> Bool
interpretacion2 "P" [x] = acuatica x
interpretacion2 "Q" [x] = terrestre x
interpretacion2 _ _      = False

```

Tomando como universo todas las entidades, menos la que denotamos Inespecífico, se tienen las siguientes evaluaciones

```

-- | Evaluaciones
-- >>> valor (take 26 entidades) interpretacion1 asignacion formula1
-- False
-- >>> valor (take 26 entidades) interpretacion2 asignacion formula1
-- True

```

Por ahora siempre hemos establecido propiedades, pero podríamos haber definido relaciones binarias, ternarias, ..., n-arias.

## 2.4. Términos funcionales

En la sección anterior todos los términos han sido variables. Ahora consideraremos cualquier término.

**Definición 2.4.1.** Son **términos** en un lenguaje de primer orden:

1. Variables
2. Constantes
3.  $f(t_1, \dots, t_n)$  si  $t_i$  son términos  $\forall i = 1, \dots, n$

Definimos un tipo de dato para los términos que serán la base para la definición de aquellas fórmulas de la lógica de primer orden que no están compuestas sólo por variables.

```

data Termino = Var Variable | Ter Nombre [Termino]
  deriving (Eq,Ord,Generic)

```

Algunos ejemplos de variables como términos

```
tx, ty, tz :: Termino
tx = Var x
ty = Var y
tz = Var z
tu = Var u
```

Como hemos introducido, también tratamos con constantes, por ejemplo:

```
a, b, c, cero :: Termino
a    = Ter "a" []
b    = Ter "b" []
c    = Ter "c" []
cero = Ter "cero" []
```

Para mostrarlo por pantalla de manera comprensiva, definimos su representación.

```
-- | Ejemplo
-- >>> Ter "f" [tx,ty]
-- f[x,y]

instance Show Termino where
  show (Var v)      = show v
  show (Ter c []) = c
  show (Ter f ts) = f ++ show ts

instance Out Termino where
  doc      = text . show
  docPrec _ = doc
```

La propiedad (`esVariable t`) se verifica si el término `t` es una variable.

```
-- | Ejemplos
-- >>> esVariable tx
-- True
-- >>> esVariable (Ter "f" [tx,ty])
-- False
esVariable :: Termino -> Bool
esVariable (Var _) = True
esVariable _      = False
```

Ahora, creamos el tipo de dato `Form` de manera análoga a como lo hicimos en la sección anterior, pero en este caso considerando cualquier término.

```
data Form = Atom Nombre [Termino]
          | Ig Termino Termino
          | Neg Form
          | Impl Form Form
          | Equiv Form Form
          | Conj [Form]
          | Disy [Form]
          | PTodo Variable Form
          | Ex Variable Form
deriving (Eq,Ord,Generic)
```

Algunos ejemplos de fórmulas son

```
formula2, formula3 :: Form
formula2 = PTodo x (PTodo y (Impl (Atom "R" [tx,ty])
                                   (Ex z (Conj [Atom "R" [tx,tz],
                                                Atom "R" [tz,ty]]))))
formula3 = Impl (Atom "R" [tx,ty])
              (Ex z (Conj [Atom "R" [tx,tz], Atom "R" [tz,ty]]))
```

Y procedemos análogamente a la sección anterior, definiendo la representación de fórmulas por pantalla.

```
instance Show Form where
  show (Atom r []) = r
  show (Atom r ts) = r ++ show ts
  show (Ig t1 t2) = show t1 ++ "≡" ++ show t2
  show (Neg f) = '¬': show f
  show (Impl f1 f2) = "(" ++ show f1 ++ "⇒" ++ show f2 ++ ")"
  show (Equiv f1 f2) = "(" ++ show f1 ++ "⇔" ++ show f2 ++ ")"
  show (Conj []) = "verdadero"
  show (Conj [f]) = show f
  show (Conj (f:fs)) = "(" ++ show f ++ "∧" ++ show (Conj fs) ++ ")"
  show (Disy []) = "falso"
  show (Disy [f]) = show f
  show (Disy (f:fs)) = "(" ++ show f ++ "∨" ++ show (Disy fs) ++ ")"
  show (PTodo v f) = "∀" ++ show v ++ (' ': show f)
  show (Ex v f) = "∃" ++ show v ++ (' ': show f)

instance Out Form where
  doc = text . show
  docPrec _ = doc
```

Quedando las fórmulas ejemplo antes definidas de la siguiente manera

```
-- | Ejemplos
-- >>> formula2
--  $\forall x \forall y (R[x,y] \implies \exists z (R[x,z] \wedge R[z,y]))$ 
-- >>> formula3
--  $(R[x,y] \implies \exists z (R[x,z] \wedge R[z,y]))$ 
```

Previamente hemos definido `InterpretacionR`, ahora para la interpretación de los símbolos funcionales se define un nuevo tipo de dato, `InterpretacionF`.

```
type InterpretacionF a = String -> [a] -> a
```

Para interpretar las fórmulas, se necesita primero una interpretación del valor en los términos.

**Definición 2.4.2.** Dada una estructura  $\mathcal{I} = (U, I)$  de  $L$  y una asignación  $A$  en  $\mathcal{I}$ , se define la **función de evaluación de términos**  $\mathcal{I}_A : \text{Term}(L) \rightarrow U$  por

$$\mathcal{I}_A(t) = \begin{cases} I(c), & \text{si } t \text{ es una constante } c \\ A(x), & \text{si } t \text{ es una variable } x \\ I(f)(\mathcal{I}_A(t_1), \dots, \mathcal{I}_A(t_n)), & \text{si } t \text{ es } f(t_1, \dots, t_n) \end{cases}$$

*Nota 2.4.1.*  $\mathcal{I}_A$  se lee “el valor de  $t$  en  $\mathcal{I}$  respecto de  $A$ ”.

El valor de `(valorT i a t)` es el valor del término  $t$  en la interpretación  $i$  respecto de la asignación  $a$ .

```
valorT :: InterpretacionF a -> Asignacion a -> Termino -> a
valorT i a (Var v)      = a v
valorT i a (Ter f ts) = i f (map (valorT i a) ts)
```

Para evaluar ejemplos de evaluación de términos usaremos la siguiente interpretación de los símbolos de función

```
interpretacionF1 :: String -> [Int] -> Int
interpretacionF1 "cero" []      = 0
interpretacionF1 "s"  [i]      = succ i
interpretacionF1 "mas" [i,j]    = i + j
interpretacionF1 "por" [i,j]    = i * j
interpretacionF1 _ _          = 0
```

y la siguiente asignación

```
asignacion1 :: Variable -> Int
asignacion1 _ = 0
```



Con ello, se tiene

```
-- | Evaluaciones
-- >>> let t1 = Ter "cero" []
-- >>> valorT interpretacionF1 asignacion1 t1
-- 0
-- >>> let t2 = Ter "s" [t1]
-- >>> t2
-- s[cero]
-- >>> valorT interpretacionF1 asignacion1 t2
-- 1
-- >>> let t3 = Ter "mas" [t2,t2]
-- >>> t3
-- mas[s[cero],s[cero]]
-- >>> valorT interpretacionF1 asignacion1 t3
-- 2
-- >>> let t4 = Ter "por" [t3,t3]
-- >>> t4
-- por[mas[s[cero],s[cero]],mas[s[cero],s[cero]]]
-- >>> valorT interpretacionF1 asignacion1 t4
-- 4
```

Definimos el tipo de dato Interpretación como un par formado por las interpretaciones de los símbolos de relación y la de los símbolos funcionales.

```
type Interpretacion a = (InterpretacionR a, InterpretacionF a)
```

**Definición 2.4.3.** Dada una estructura  $\mathcal{I} = (U, I)$  de  $L$  y una asignación  $A$  sobre  $\mathcal{I}$ , se define la **función evaluación de fórmulas**  $\mathcal{I}_A : \text{Form}(L) \rightarrow \text{Bool}$  por

- Si  $F$  es  $t_1 = t_2$ ,  $\mathcal{I}_A(F) = H_{=}( \mathcal{I}_A(t_1), \mathcal{I}_A(t_2) )$
- Si  $F$  es  $P(t_1, \dots, t_n)$ ,  $\mathcal{I}_A(F) = H_{I(P)}( \mathcal{I}_A(t_1), \dots, \mathcal{I}_A(t_n) )$
- Si  $F$  es  $\neg G$ ,  $\mathcal{I}_A(F) = H_{\neg}( \mathcal{I}_A(G) )$
- Si  $F$  es  $G * H$ ,  $\mathcal{I}_A(F) = H_{*}( \mathcal{I}_A(G), \mathcal{I}_A(H) )$
- Si  $F$  es  $\forall x G$ ,

$$\mathcal{I}_A(F) = \begin{cases} 1, & \text{si para todo } u \in U \text{ se tiene } \mathcal{I}_{A[x/u]} = 1 \\ 0, & \text{en caso contrario.} \end{cases}$$

- Si  $F$  es  $\exists x G$ ,

$$\mathcal{I}_A(F) = \begin{cases} 1, & \text{si existe algún } u \in U \text{ tal que } \mathcal{I}_{A[x/u]} = 1 \\ 0, & \text{en caso contrario.} \end{cases}$$

Definimos una función que determine el valor de una fórmula. Dicha función la denotamos por  $(\text{valorF } u \text{ (iR, iF) } a \text{ } f)$ , en la que  $u$  denota el universo,  $iR$  es la interpretación de los símbolos de relación,  $iF$  es la interpretación de los símbolos de función,  $a$  la asignación y  $f$  la fórmula.

```
valorF :: Eq a => Universo a -> Interpretacion a -> Asignacion a
        -> Form -> Bool
valorF u (iR,iF) a (Atom r ts) =
  iR r (map (valorT iF a) ts)
valorF u (_,iF) a (Ig t1 t2) =
  valorT iF a t1 == valorT iF a t2
valorF u i a (Neg g) =
  not (valorF u i a g)
valorF u i a (Impl f1 f2) =
  valorF u i a f1 <= valorF u i a f2
valorF u i a (Equiv f1 f2) =
  valorF u i a f1 == valorF u i a f2
valorF u i a (Conj fs) =
  all (valorF u i a) fs
valorF u i a (Disy fs) =
  any (valorF u i a) fs
valorF u i a (PTodo v g) =
  and [valorF u i (sustituye a v d) g | d <- u]
valorF u i a (Ex v g) =
  or  [valorF u i (sustituye a v d) g | d <- u]
```

Para construir un ejemplo tenemos que interpretar los elementos de una fórmula. Definimos las fórmulas 4 y 5, aunque emplearemos en el ejemplo sólo la formula4.

```
-- | Ejemplos
-- >>> formula4
--  $\exists x R[\text{cero}, x]$ 
-- >>> formula5
--  $(\forall x P[x] \implies \forall y Q[x, y])$ 
formula4, formula5 :: Form
formula4 = Ex x (Atom "R" [cero, tx])
formula5 = Impl (PTodo x (Atom "P" [tx])) (PTodo y (Atom "Q" [tx, ty]))
```

En este caso tomamos como universo  $U$  los números naturales. Interpretamos  $R$  como la desigualdad  $<$ . Es decir, vamos a comprobar si es cierto que existe un número natural mayor que el 0. Por tanto, la interpretación de los símbolos de relación es

```
interpretacionR1 :: String -> [Int] -> Bool
interpretacionR1 "R" [x,y] = x < y
interpretacionR1 _ _       = False
```

En este caso se tiene las siguientes evaluaciones

```
-- | Evaluaciones
-- >>> valorF [0..] (interpretacionR1,interpretacionF1) asignacion1 formula4
-- True
```

*Nota 2.4.2.* Haskell es perezoso, así que podemos utilizar un universo infinito. Haskell no hace cálculos innecesarios; es decir, para cuando encuentra un elemento que cumple la propiedad.

Dada una fórmula  $F$  de  $L$  se tienen las siguientes definiciones:

**Definición 2.4.4.** ■ Un **modelo** de una fórmula  $F$  es una interpretación para la que  $F$  es verdadera.

- Una fórmula  $F$  es **válida** si toda interpretación es modelo de la fórmula.
- Una fórmula  $F$  es **satisfacible** si existe alguna interpretación para la que sea verdadera.
- Una fórmula es **insatisfacible** si no tiene ningún modelo.

### 2.4.1. Generadores

Pendiente de revisión.

Para poder emplear el sistema de comprobación QuickCheck, necesitamos poder generar elementos aleatorios de los tipos de datos creados hasta ahora.

```
module Generadores where
import PFH
import Modelo
import LPH
import Dominio
import Test.QuickCheck
import Control.Monad
```

#### Generador de Nombres

```
abecedario :: Nombre
abecedario = "abcdefghijklmnopqrstuvwxyz"

genLetra :: Gen Char
genLetra = elements abecedario
```

Ejemplo de generación de letras

```
ghci> sample genLetra
'w'
'r'
'l'
'o'
'u'
'z'
'f'
'x'
'k'
'q'
'b'
```

```
genNombre :: Gen Nombre
genNombre = liftM (take 1) (listOf1 genLetra)
```

Se puede definir genNombre como sigue

```
genNombre2 :: Gen Nombre
genNombre2 = do
  c <- elements ['a'..'z']
  return [c]
```

Ejemplo de generación de nombres

```
ghci> sample genNombre2
"z"
"u"
"j"
"h"
"v"
"w"
"v"
"b"
"e"
"d"
"s"
```

## Generador de Índices

```
genNumero :: Gen Int
genNumero = choose (0,100)

genIndice :: Gen Indice
genIndice = liftM (take 1) (listOf1 genNumero)
```

### Ejemplo

```
ghci> sample genIndice
[98]
[62]
[50]
[89]
[97]
[6]
[14]
[87]
[14]
[92]
[1]
```

### Generador de variables

```
generaVariable :: Gen Variable
generaVariable = liftM2 Variable (genNombre) (genIndice)

instance Arbitrary (Variable) where
    arbitrary = generaVariable
```

### Ejemplo

```
ghci> sample generaVariable
q10
e5
m97
n92
h15
a52
c58
s74
t30
g78
i75
```

### Generador de Fórmulas

```
instance Arbitrary (Formula) where
    arbitrary = sized formula
        where
            formula 0 = liftM2 Atomo genNombre (listOf generaVariable)
            formula n = oneof [liftM Negacion generaFormula,
```

```

liftM2 Implica generaFormula generaFormula,
liftM2 Equivalente generaFormula generaFormula,
liftM Conjunction (listOf generaFormula),
liftM Disyuncion (listOf generaFormula),
liftM2 ParaTodo generaVariable generaFormula,
liftM2 Existe generaVariable generaFormula]

where
    generaFormula = formula (n-1)

```

### Generador de Términos

```

instance Arbitrary (Termino) where
    arbitrary = sized termino
    where
        termino 0 = liftM Var generaVariable
        termino n = liftM2 Ter genNombre (listOf generaTermino)
            where
                generaTermino = termino (n-1)

```

### 2.4.2. Otros conceptos de la lógica de primer orden

Las funciones `varEnTerm` y `varEnTerms` devuelven las variables que aparecen en un término o en una lista de ellos.

```

-- Ejemplos
-- >>> let t1 = Ter "f" [tx,a]
-- >>> varEnTerm t1
-- [x]
-- >>> let t2 = Ter "g" [tx,a,ty]
-- >>> varEnTerm t2
-- [x,y]
-- >>> varEnTerms [t1,t2]
-- [x,y]
varEnTerm :: Termino -> [Variable]
varEnTerm (Var v)    = [v]
varEnTerm (Ter _ ts) = varEnTerms ts

varEnTerms :: [Termino] -> [Variable]
varEnTerms = nub . concatMap varEnTerm

```

*Nota 2.4.3.* La función `nub xs` elimina elementos repetidos en una lista `xs`. Se encuentra en el paquete `Data.List`.

*Nota 2.4.4.* Se emplea un tipo de recursión cruzada entre funciones. Las funciones se llaman la una a la otra.

La función `varEnForm` devuelve una lista de las variables que aparecen en una fórmula.

```
-- | Ejemplos
-- >>> formula2
--  $\forall x \forall y (R[x,y] \implies \exists z (R[x,z] \wedge R[z,y]))$ 
-- >>> varEnForm formula2
-- [x,y,z]
-- >>> formula3
--  $(R[x,y] \implies \exists z (R[x,z] \wedge R[z,y]))$ 
-- >>> varEnForm formula3
-- [x,y,z]
-- >>> formula4
--  $\exists x R[\text{cero},x]$ 
-- >>> varEnForm formula4
-- [x]
varEnForm :: Form -> [Variable]
varEnForm (Atom _ ts)    = varEnTerms ts
varEnForm (Ig t1 t2)     = nub (varEnTerm t1 ++ varEnTerm t2)
varEnForm (Neg f)        = varEnForm f
varEnForm (Impl f1 f2)   = varEnForm f1 'union' varEnForm f2
varEnForm (Equiv f1 f2)  = varEnForm f1 'union' varEnForm f2
varEnForm (Conj fs)      = nub (concatMap varEnForm fs)
varEnForm (Disy fs)      = nub (concatMap varEnForm fs)
varEnForm (PTodo x f)    = varEnForm f
varEnForm (Ex x f)       = varEnForm f
```

**Definición 2.4.5.** Una variable es **libre** en una fórmula si no tiene ninguna aparición ligada a un cuantificador existencial o universal. ( $\forall x, \exists x$ )

La función `(variablesLibres f)` devuelve las variables libres de la fórmula `f`.

```
-- | Ejemplos
-- >>> formula2
--  $\forall x \forall y (R[x,y] \implies \exists z (R[x,z] \wedge R[z,y]))$ 
-- >>> variablesLibres formula2
-- []
-- >>> formula3
--  $(R[x,y] \implies \exists z (R[x,z] \wedge R[z,y]))$ 
-- >>> variablesLibres formula3
-- [x,y]
-- >>> formula4
--  $\exists x R[\text{cero},x]$ 
-- >>> variablesLibres formula4
```

```

-- []
variablesLibres :: Form -> [Variable]
variablesLibres (Atom _ ts) =
    varEnTerms ts
variablesLibres (Ig t1 t2) =
    varEnTerm t1 'union' varEnTerm t2
variablesLibres (Neg f) =
    variablesLibres f
variablesLibres (Impl f1 f2) =
    variablesLibres f1 'union' variablesLibres f2
variablesLibres (Equiv f1 f2) =
    variablesLibres f1 'union' variablesLibres f2
variablesLibres (Conj fs) =
    nub (concatMap variablesLibres fs)
variablesLibres (Disy fs) =
    nub (concatMap variablesLibres fs)
variablesLibres (PTodo x f) =
    delete x (variablesLibres f)
variablesLibres (Ex x f) =
    delete x (variablesLibres f)

```

**Definición 2.4.6.** Una variable  $x$  está **ligada** en una fórmula cuando tiene una aparición de la forma  $\forall x$  o  $\exists x$ .

**Definición 2.4.7.** Una **fórmula abierta** es una fórmula con variables libres.

La función (`formulaAbierta f`) determina si una fórmula dada es abierta.

```

-- Ejemplos
-- >>> formula3
-- (R[x,y] ==> ∃z (R[x,z] ∧ R[z,y]))
-- >>> formulaAbierta formula3
-- True
-- >>> formula2
-- ∀x ∀y (R[x,y] ==> ∃z (R[x,z] ∧ R[z,y]))
-- >>> formulaAbierta formula2
-- False
formulaAbierta :: Form -> Bool
formulaAbierta = not . null . variablesLibres

```



# Capítulo 3

## Deducción natural

En este capítulo se pretende implementar la deducción natural de la lógica de primer orden en Haskell. El contenido de este capítulo se encuentra en el módulo DNH.

```
module DNH where
import LPH
import Data.List
import Text.PrettyPrint
import Text.PrettyPrint.GenericPretty
```

### 3.1. Sustitución

**Definición 3.1.1.** Una **sustitución** es una aplicación  $S : Variable \rightarrow Termino$ .

*Nota 3.1.1.*  $[x_1/t_1, x_2/t_2, \dots, x_n/t_n]$  representa la sustitución

$$S(x) = \begin{cases} t_i, & \text{si } x \text{ es } x_i \\ x, & \text{si } x \notin \{x_1, \dots, x_n\} \end{cases}$$

En la lógica de primer orden, a la hora de emplear el método de tableros, es necesario sustituir las variables ligadas por términos. Por lo tanto, requerimos de la definición de un nuevo tipo de dato para las sustituciones.

```
type Sust = [(Variable, Termino)]
```

Sería interesante comparar la representación de sustituciones mediante diccionarios con la librería `Data.Map`

Este nuevo tipo de dato es una asociación de la variable con el término mediante pares. Denotamos el elemento identidad de la sustitución como identidad



**Definición 3.1.2.**  $t[x_1/t_1, \dots, x_n/t_n]$  es el término obtenido sustituyendo en  $t$  las apariciones de  $x_i$  por  $t_i$ .

**Definición 3.1.3.** La extensión de la sustitución a términos es la aplicación  $S : \text{Term}(L) \rightarrow \text{Term}(L)$  definida por

$$S(t) = \begin{cases} c, & \text{si } t \text{ es una constante } c \\ S(x), & \text{si } t \text{ es una variable } x \\ f(S(t_1), \dots, S(t_n)), & \text{si } t \text{ es } f(t_1, \dots, t_n) \end{cases}$$

Ahora, aplicando recursión entre funciones, podemos hacer sustituciones basándonos en los términos mediante las funciones `(susTerm xs t)` y `(susTerms sust ts)`.

```
susTerm :: Sust -> Termino -> Termino
susTerm s (Var y)    = sustituyeVar s y
susTerm s (Ter f ts) = Ter f (susTerms s ts)

susTerms :: Sust -> [Termino] -> [Termino]
susTerms = map . susTerm
```

Por ejemplo,

```
-- | Ejemplos
-- >>> susTerm [(x,ty)] tx
-- y
-- >>> susTerms [(x,ty),(y,tx)] [tx,ty]
-- [y,x]
```

**Definición 3.1.4.**  $F[x_1/t_1, \dots, x_n/t_n]$  es la fórmula obtenida sustituyendo en  $F$  las apariciones libres de  $x_i$  por  $t_i$ .

**Definición 3.1.5.** La extensión de  $S$  a fórmulas es la aplicación  $S : \text{Form}(L) \rightarrow \text{Form}(L)$  definida por

$$S(F) = \begin{cases} P(S(t_1), \dots, S(t_n)), & \text{si } F \text{ es la fórmula atómica } P(t_1, \dots, t_n) \\ S(t_1) = S(t_2), & \text{si } F \text{ es la fórmula } t_1 = t_2 \\ \neg(S(G)), & \text{si } F \text{ es } \neg G \\ S(G) * S(H), & \text{si } F \text{ es } G * H \\ (Qx)(S_x(G)), & \text{si } F \text{ es } (Qx)G \end{cases}$$

donde  $S_x$  es la sustitución definida por

$$S_x(y) = \begin{cases} x, & \text{si } y \text{ es } x \\ S(y), & \text{si } y \text{ es distinta de } x \end{cases}$$

Definimos (`sustitucionForm s f`), donde `s` representa la sustitución y `f` la fórmula.

```
sustitucionForm :: Sust -> Form -> Form
sustitucionForm s (Atom r ts) =
  Atom r (susTerms s ts)
sustitucionForm s (Ig t1 t2) =
  Ig (susTerm s t1) (susTerm s t2)
sustitucionForm s (Neg f) =
  Neg (sustitucionForm s f)
sustitucionForm s (Impl f1 f2) =
  Impl (sustitucionForm s f1) (sustitucionForm s f2)
sustitucionForm s (Equiv f1 f2) =
  Equiv (sustitucionForm s f1) (sustitucionForm s f2)
sustitucionForm s (Conj fs) =
  Conj (sustitucionForms s fs)
sustitucionForm s (Disy fs) =
  Disy (sustitucionForms s fs)
sustitucionForm s (PTodo v f) =
  PTodo v (sustitucionForm s' f)
  where s' = [x | x <- s, fst x /= v]
sustitucionForm s (Ex v f) =
  Ex v (sustitucionForm s' f)
  where s' = [x | x <- s, fst x /= v]
```

Por ejemplo,

```
-- | Ejemplos
-- >>> formula3
-- (R[x,y] ==> ∃z (R[x,z] ∧ R[z,y]))
-- >>> sustitucionForm [(x,ty)] formula3
-- (R[y,y] ==> ∃z (R[y,z] ∧ R[z,y]))
```

Se puede generalizar a una lista de fórmulas mediante la función (`sustitucionForms s fs`). La hemos necesitado en la definición de la función anterior, pues las conjunciones y disyunciones trabajan con listas de fórmulas.

```
sustitucionForms :: Sust -> [Form] -> [Form]
sustitucionForms s = map (sustitucionForm s)
```

Nos podemos preguntar si la sustitución conmuta con la composición. Para ello definimos la función (`composicion s1 s2`)

```
composicion :: Sust -> Sust -> Sust
composicion s1 s2 =
  hacerApropiada [(y,susTerm s1 y') | (y,y') <- s2] ++
  [x | x <- s1, fst x 'notElem' dominio s2]
```

Por ejemplo,

```
-- | Ejemplos
-- >>> composicion [(x,tx)] [(y,ty)]
-- [(x,x)]
-- >>> composicion [(x,tx)] [(x,ty)]
-- [(x,y)]
```

```
composicionConmutativa :: Sust -> Sust -> Bool
composicionConmutativa s1 s2 =
  composicion s1 s2 == composicion s2 s1
```

Y comprobando con QuickCheck que no lo es

```
ghci> quickCheck composicionConmutativa
*** Failed! Falsifiable (after 3 tests and 1 shrink):
[(i3,n)]
[(c19,i)]
```

Un contraejemplo más claro es

```
composicion [(x,tx)] [(y,ty)] == [(x,x)]
composicion [(y,ty)] [(x,tx)] == [(y,y)]
```

*Nota 3.1.2.* Las comprobaciones con QuickCheck emplean código del módulo Generadores.

**Definición 3.1.6.** Una sustitución se denomina **libre para una fórmula** cuando todas las apariciones de variables introducidas por la sustitución en esa fórmula resultan libres.

Un ejemplo de una sustitución que no es libre

```
-- | Ejemplo
-- >>> let f1 = Ex x (Atom "R" [tx,ty])
-- >>> f1
-- ∃x R[x,y]
-- >>> variablesLibres f1
-- [y]
-- >>> sustitucionForm [(y,tx)] f1
-- ∃x R[x,x]
-- >>> variablesLibres (sustitucionForm [(y,tx)] f1)
-- []
```

Un ejemplo de una sustitución libre

```
-- | Ejemplo
-- >>> formula5
-- ( $\forall x P[x] \implies \forall y Q[x,y]$ )
-- >>> variablesLibres formula5
-- [x]
-- >>> sustitucionForm [(x,tz)] formula5
-- ( $\forall x P[x] \implies \forall y Q[z,y]$ )
-- >>> variablesLibres (sustitucionForm [(x,tz)] formula5)
-- [z]
```

## 3.2. Sustitucion mediante diccionarios

En esta sección definiremos las sustituciones ,de una manera alternativa, mediante la librería `Data.Map`.

```
module SustDiccionario where
import LPH
import Data.List
import Data.Map (Map)
import qualified Data.Map as M
import Text.PrettyPrint
import Text.PrettyPrint.GenericPretty
```

Debido a que muchas funciones de esta librería coinciden con funciones definidas en `prelude`, ésta se suele importar `qualified`.

No compatible con nuestra definición de término, hay que adaptarlo

## 3.3. Reglas de deducción natural

La deducción natural está compuesta de una serie de reglas a través de las cuales, partiendo de una fórmula o un conjunto de ellas, conseguimos deducir otra u otras fórmulas.

Se definen los átomos `p,q` y `r` para comodidad en los ejemplos.

```
p = Atom "p" []
q = Atom "q" []
r = Atom "r" []
```

A continuación se introducen los tipos de datos necesarios para la definición de las reglas de la deducción natural:

```

data Reglas = Suponer Form
    -- Reglas de la conjunción:
    | IntroConj Form Form
    | ElimConjI Form
    | ElimConjD Form
    -- Reglas de la negación:
    | ElimDobleNeg Form
    | IntroDobleNeg Form
    | ElimNeg Form
    | IntroNeg Form
    | ElimFalso Form
    -- Reglas del condicional:
    | ElimImpl Form Form
    | IntroImpl Form Form
    -- Modus Tollens:
    | MT Form Form
    -- Reglas de la disyunción:
    | IntroDisyI Form Form
    | IntroDisyD Form Form
    | ElimDisy Form Form
    -- Reglas de la bicondicional:
    | IntroEquiv Form
    | ElimEquivI Form
    | ElimEquivD Form
    -- Reducción al absurdo:
    | RedAbsurdo Form
    -- Regla del tercio excluido:
    | TercioExcl Form
    -- Reglas del cuantificador existencial y universal:
    | ElimUniv Form Sust
    | IntroUniv Sust Form
    deriving Show

```

Cuando elaboramos una deducción a partir de una serie de premisas trabajaremos con una lista de “cosas conocidas” y otra de “cosas supuestas”. Para ello definimos el tipo de dato `Deduccion` de la siguiente forma:

```

data Deduccion = D [Form] [Form] [Reglas]
    deriving Show

```

Finalmente, se define como un átomo el elemento contradicción, pues nos será necesario en la definición de algunas reglas.

```

contradiccion :: Form
contradiccion = Atom "⊥" []

```

Implementamos en Haskell un par de funciones auxiliares (`quita xs ys`) que elimina todos los elementos `xs` de la lista `ys`, y (`elemMap xs ys`) que determina si una lista está contenida en la otra.

```
quita :: [Form] -> [Form] -> [Form]
quita [] ys = ys
quita (x:xs) ys = quita xs (delete x ys)

elemMap :: [Form] -> [Form] -> Bool
elemMap xs ys = all ('elem' ys) xs
```

Por ejemplo,

```
-- | Ejemplos
-- >>> quita [p,q] [p,q,contradiccion]
-- [⊥]
-- >>> elemMap [p,q] [p,q,contradiccion]
-- True
-- >>> elemMap [contradiccion,q] [p,contradiccion]
-- False
```

En las posteriores subsecciones se va a definir la función (`verifica d`), donde `d` será un elemento del tipo de dato `Deducccion`, y que pretende determinar si un proceso elaborado por deducción natural es correcto.

```
verifica :: Deducccion -> Bool
```

Los primeros casos en la función `verifica` serán el básico, es decir, en el que determinaremos que el proceso deductivo es correcto, y la regla antes definida en el tipo de dato `Reglas` como `Suponer`, cuya función va a ser incluir una fórmula en la lista de las suposiciones. Lo implementamos en la función `verifica`.

```
verifica (D pr [] []) = True
verifica (D pr sp []) = error "Quedan supuestos"
verifica (D pr sp ((Suponer f):rs)) = verifica (D pr (f:sp) rs)
```

### 3.3.1. Reglas de la conjunción

- Regla de la introducción de la conjunción:

$$\frac{F \quad G}{F \wedge G}$$

Cuya implementación en la función `verifica` es:



```

verifica (D pr sp ((IntroConj f g):rs))
  | elemMap [f,g] (pr++sp) =
    verifica (D ((Conj [f,g]):pr) (quita [f,g] sp) rs)
  | otherwise = error "No se puede aplicar IntroConj"

```

- Regla de la eliminación de la conjunción:

$$\frac{F_1 \wedge \cdots \wedge F_n}{F_1} \text{ y } \frac{F_1 \wedge \cdots \wedge F_n}{F_n}.$$

Que se implementan de la siguiente forma:

```

verifica (D pr sp ((ElimConjI f@(Conj (f1:fs))):rs))
  | elem f (pr++sp) = verifica (D (f1:pr) (delete f sp) rs)
  | otherwise = error "No se puede aplicar ElimConjI"

verifica (D pr sp ((ElimConjD f@(Conj fs)):rs))
  | elem f (pr++sp) = verifica (D ((last fs):pr) (delete f sp) rs)
  | otherwise = error "No se puede aplicar ElimConjD"

```

*Nota 3.3.1.* La función error permite mostrar un mensaje por pantalla, en el que podemos aclarar la razón del error.

### 3.3.2. Reglas de eliminación del condicional

- Regla de la eliminación del condicional:

$$\frac{F \quad F \rightarrow G}{G}$$

```

verifica (D pr sp ((ElimImpl f1 form@(Impl f2 g)):rs))
  | c = verifica (D (g:pr) (quita [f1,form] sp) rs)
  | otherwise = error "No se puede aplicar la ElimImpl"
where
  c = elemMap [f1,form] (pr++sp) && f1 == f2

```

- Regla de introducción del condicional:

$$\frac{\begin{bmatrix} F \\ \vdots \\ G \end{bmatrix}}{F \rightarrow G}$$

```

verifica (D pr sp ((IntroImpl f g):rs))
  | elem f sp && elem g pr =

```

```
verifica (D ((Impl f g):pr) (delete f sp) rs)
| otherwise = error "No se puede aplicar IntroImpl"
```

### 3.3.3. Reglas de la disyunción

- Reglas de la introducción de la disyunción:

$$\frac{F}{F \vee G} \text{ y } \frac{G}{F \vee G}$$

```
verifica (D pr sp ((IntroDisyI f g):rs))
| elem f (pr++sp) =
    verifica (D ((Disy [f,g]):pr) (delete f sp) rs)
| otherwise = error "No se puede aplicar IntroDisyI"

verifica (D pr sp ((IntroDisyD f g):rs))
| elem g (pr++sp) =
    verifica (D ((Disy [f,g]):pr) (delete g sp) rs)
| otherwise = error "No se puede aplicar IntroDisyD"
```

- Regla de la eliminación de la disyunción:

$$\frac{F \vee G \quad \begin{array}{c} \left[ \begin{array}{c} F \\ \vdots \\ H \end{array} \right] \quad \left[ \begin{array}{c} G \\ \vdots \\ H \end{array} \right] }{H}$$

```
verifica (D pr sp ((ElimDisy (Disy [f,g]) h):rs))
| c = verifica (D (h:pr) (quita [f,g] sp) rs)
| otherwise = error "No se puede aplicar ElimDisy"
where
    c = elemMap [f,g] sp
        && elem (Conj [f,g]) pr
        && elem h pr
        && elem h (delete h pr)
```

### 3.3.4. Reglas de la negación

- Regla de eliminación de lo falso:

$$\frac{\perp}{F}$$

```
verifica (D pr sp ((ElimFalso f):rs))
| elem contradiccion pr =
```

```
verifica (D (f:(delete contradiccion pr)) sp rs)
| otherwise = error "No se puede aplicar ElimFalso"
```

- Regla de eliminación de la negación

$$\frac{F \quad \neg F}{\perp}$$

```
verifica (D pr sp ((ElimNeg f):rs))
| elem f (pr++sp) && elem (Neg f) (pr++sp) =
    verifica (D (contradiccion:pr) (quita [f,Neg f] sp) rs)
| otherwise = error "No se puede aplicar ElimNeg"
```

### 3.3.5. Reglas del bicondicional

- Regla de introducción del bicondicional:

$$\frac{F \rightarrow G \quad G \rightarrow F}{F \leftrightarrow G}$$

```
verifica (D pr sp ((IntroEquiv f1@(Impl f g)):rs))
| elemMap [f1,Impl g f] (pr++sp) =
    verifica (D ((Equiv f g):pr) (quita [f1,Impl g f] sp) rs)
| otherwise = error "No se puede aplicar IntroEquiv"
```

- Reglas de la eliminación del bicondicional:

$$\frac{F \leftrightarrow G}{F \rightarrow G} \text{ y } \frac{F \leftrightarrow G}{G \rightarrow F}$$

```
verifica (D pr sp ((ElimEquivI f1@(Equiv f g)):rs))
| elem f1 (pr++sp) =
    verifica (D ((Impl f g):pr) (delete f1 sp) rs)
| otherwise = error "No se puede aplicar ElimEquivI"
verifica (D pr sp ((ElimEquivD f1@(Equiv f g)):rs))
| elem f1 (pr++sp) =
    verifica (D ((Impl g f):pr) (delete f1 sp) rs)
| otherwise = error "No se puede aplicar ElimEquivD"
```

### 3.3.6. Regla derivada de Modus Tollens(MT)

- Regla derivada de modus Tollens:

$$\frac{F \rightarrow G \quad \neg G}{\neg F}$$

```

verifica (D pr sp ((MT f1@(Impl f g) (Neg g1)):rs))
  | elemMap [f1,Neg g1] (pr++sp) && g == g1 =
    verifica (D ((Neg f):pr) (quita [f1,Neg g1] sp) rs)
  | otherwise = error "No se puede aplicar MT"

```

### 3.3.7. Reglas de la doble negación

- Regla de eliminación de la doble negación:

$$\frac{\neg\neg F}{F}$$

```

verifica (D pr sp ((ElimDobleNeg form@(Neg (Neg f))):rs))
  | elem form (pr++sp) = verifica (D (f:pr) (delete form sp) rs)
  | otherwise = error "No se puede aplicar ElimDobleNeg"

```

- Regla de la introducción de la doble negación:

$$\frac{F}{\neg\neg F}$$

```

verifica (D pr sp ((IntroDobleNeg f):rs))
  | elem f (pr++sp) =
    verifica (D ((Neg (Neg f)):pr) (delete f sp) rs)
  | otherwise = error "No se puede aplicar la IntroDobleNeg"

```

### 3.3.8. Regla de Reducción al absurdo

- Regla de reducción al absurdo:

$$\frac{\begin{array}{c} \neg F \\ \vdots \\ \perp \end{array}}{F}$$

```

verifica (D pr sp ((RedAbsurdo (Neg f)):rs))
  | elem (Neg f) sp && elem contradiccion pr =
    verifica (D (f:(delete contradiccion pr))
      (delete (Neg f) sp) rs)
  | otherwise = error "No se puede aplicar RedAbsurdo"

```

### 3.3.9. Ley del tercio excluido

- Ley del tercio excluido:

$$\overline{F \vee \neg F}$$

```
verifica (D pr sp ((TercioExcl f):rs)) =
  verifica (D ((Disy [f,Neg f]):pr) sp rs)
```

### 3.3.10. Reglas del cuantificador universal

- Regla de eliminación del cuantificador universal:

$$\frac{\forall x F}{F[x/t]} \text{ donde } [x/t] \text{ es libre para } F$$

```
verifica (D pr sp ((ElimUniv f@(PTodo v g) s@([(v',t)])):rs))
  | elem f (pr++sp) && (v==v') && elem v (variablesLibres f) =
    verifica (D ((sustitucionForm s g):pr) sp rs)
  | otherwise = error "No se puede aplicar ElimUniv"
```

- Regla de la introducción del cuantificador universal:

$$\frac{\begin{bmatrix} x_0 \\ \vdots \\ F[x/x_0] \end{bmatrix}}{\forall x F}$$

donde  $x_0$  es una variable nueva, que no aparece fuera de la caja.

```
verifica (D pr sp ((IntroUniv s@([(v,t)])) f):rs))
  | elem (Atom [] [t]) sp && elem f (pr++sp) =
    verifica (D ((sustitucionForm s f):pr)
      (delete (Atom [] [t]) sp) rs)
  | otherwise = error "No se puede aplicar IntroUniv"
```

### 3.3.11. Reglas del cuantificador existencial

- Regla de introducción del cuantificador existencial:

$$\frac{F[x/t]}{\exists x F} \text{ donde } [x/t] \text{ es libre para } F$$

### 3.3.12. Ejemplos

- Ejemplo:  $\neg q \rightarrow \neg p \vdash p \rightarrow \neg\neg q$

1.  $\neg q \rightarrow \neg p$

2.  $p$

3.  $\neg\neg p$

4.  $\neg\neg q$

5.  $p \rightarrow \neg\neg q$

```
-- | Ejemplos
-- >>> let rs = [Suponer p,IntroDobleNeg p, MT (Impl (Neg q) (Neg p))
-- (Neg (Neg p)), IntroImpl p (Neg (Neg q))]]
-- >>> rs
-- [Suponer p,IntroDobleNeg p,MT (¬q⇒¬p) ¬¬p,IntroImpl p ¬¬q]
-- >>> verifica (D [Impl (Neg q) (Neg p)] [] rs)
-- True
```

# Capítulo 4

## Prueba de teoremas en lógica de predicados

Este capítulo pretende aplicar métodos de tableros para la demostración de teoremas en lógica de predicados. El contenido de este capítulo se encuentra en el módulo PTLP.

```
{-# LANGUAGE DeriveGeneric #-}
module PTLP where
import LPH
import DNH
import Data.List
import Test.QuickCheck -- Para ejemplos
import Generadores      -- Para ejemplos
import Text.PrettyPrint
import Text.PrettyPrint.GenericPretty
```

### 4.1. Unificación

**Definición 4.1.1.** Un **unificador** de dos términos  $t_1$  y  $t_2$  es una sustitución  $S$  tal que  $S(t_1) = S(t_2)$ .

```
unificadoresTerminos :: Termino -> Termino -> [Sust]
unificadoresTerminos (Var x) (Var y)
  | x == y    = [identidad]
  | otherwise = [[(x,Var y)]]
unificadoresTerminos (Var x) t =
  [[(x,t)] | x 'notElem' varEnTerm t]
unificadoresTerminos t (Var y) =
  [[(y,t)] | y 'notElem' varEnTerm t]
unificadoresTerminos (Ter f ts) (Ter g rs) =
  [u | f == g, u <- unificadoresListas ts rs]
```

El valor de `(unificadoresListas ts rs)` es un unificador de las listas de términos `ts` y `rs`; es decir, una sustitución `s` tal que si `ts = [t1, ..., tn]` y `rs = [r1, ..., rn]` entonces `s(t1) = s(r1), ..., s(tn) = s(rn)`.

```
unificadoresListas :: [Termino] -> [Termino] -> [Sust]
unificadoresListas [] [] = [identidad]
unificadoresListas [] _ = []
unificadoresListas _ [] = []
unificadoresListas (t:ts) (r:rs) =
  [composicion u1 u2
   | u1 <- unificadoresTerminos t r
   , u2 <- unificadoresListas (susTerms u1 ts) (susTerms u1 rs)]
```

Por ejemplo,

```
unificadoresListas [tx] [ty] == [(x,y)]
unificadoresListas [tx] [tx] == []
```

## 4.2. Skolem

### 4.2.1. Formas normales

**Definición 4.2.1.** Una fórmula está en **forma normal conjuntiva** si es una conjunción de disyunciones de literales.

$$(p_1 \vee \cdots \vee p_n) \wedge \cdots \wedge (q_1 \vee \cdots \vee q_m)$$

*Nota 4.2.1.* La forma normal conjuntiva es propia de la lógica proposicional. Por ello las fórmulas aquí definidas sólo se aplicaran a fórmulas sin cuantificadores.

Definimos la función `(enFormaNC f)` para determinar si una fórmula está en su forma normal conjuntiva.

```
enFormaNC :: Form -> Bool
enFormaNC (Atom _ _) = True
enFormaNC (Conj fs) = and [(literal f) || (esConj f) | f <- fs]
  where
    esConj (Disy fs) = all (literal) fs
    esConj _ = False
enFormaNC _ = False
```

Por ejemplo



```
-- | Ejemplos
-- >>> enFormaNC (Conj [p, Disy [q,r]])
-- True
-- >>> enFormaNC (Conj [Impl p r, Disy [q, Neg r]])
-- False
-- >>> enFormaNC (Conj [p, Disy [q, Neg r]])
-- True
```

Aplicando a una fórmula  $F$  el siguiente algoritmo se obtiene una forma normal conjuntiva de  $F$ .

1. Eliminar los bicondicionales usando la equivalencia

$$A \leftrightarrow B \equiv (A \rightarrow B) \wedge (B \rightarrow A)$$

2. Eliminar los condicionales usando la equivalencia

$$A \rightarrow B \equiv \neg A \vee B$$

3. Interiorizar las negaciones usando las equivalencias

$$\neg(A \wedge B) \equiv \neg A \vee \neg B$$

$$\neg(A \vee B) \equiv \neg A \wedge \neg B$$

$$\neg\neg A \equiv A$$

4. Interiorizar las disyunciones usando las equivalencias

$$A \vee (B \wedge C) \equiv (A \vee B) \wedge (A \vee C)$$

$$(A \wedge B) \vee C \equiv (A \vee C) \wedge (B \vee C)$$

Implementamos estos pasos en Haskell

Definimos la función (`elimImpEquiv f`), para obtener fórmulas equivalentes sin equivalencias ni implicaciones.

```
elimImpEquiv :: Form -> Form
elimImpEquiv (Atom f xs) =
  Atom f xs
elimImpEquiv (Ig t1 t2) =
  Ig t1 t2
```

```

elimImpEquiv (Equiv f1 f2) =
  Conj [elimImpEquiv (Impl f1 f2),
        elimImpEquiv (Impl f2 f1)]
elimImpEquiv (Impl f1 f2) =
  Disy [Neg (elimImpEquiv f1), (elimImpEquiv f2)]
elimImpEquiv (Neg f) =
  Neg (elimImpEquiv f)
elimImpEquiv (Disy fs) =
  Disy (map elimImpEquiv fs)
elimImpEquiv (Conj fs) =
  Conj (map elimImpEquiv fs)
elimImpEquiv (Ex x f) =
  Ex x (elimImpEquiv f)
elimImpEquiv (PTodo x f) =
  PTodo x (elimImpEquiv f)

```

*Nota 4.2.2.* Se aplica a fórmulas con cuantificadores pues la función será empleada en la sección de la forma de skolem.

Por ejemplo,

```

-- | Ejemplo
-- >>> elimImpEquiv (Neg (Conj [p, Impl q r]))
-- ¬(p ∧ (¬q ∨ r))

```

Interiorizamos las negaciones mediante la función (`interiorizaNeg f`).

```

interiorizaNeg :: Form -> Form
interiorizaNeg p@(Atom f xs) = p
interiorizaNeg (Equiv f1 f2) =
  Equiv (interiorizaNeg f1) (interiorizaNeg f2)
interiorizaNeg (Impl f1 f2) = Impl (interiorizaNeg f1) (interiorizaNeg f2)
interiorizaNeg (Neg (Disy fs)) = Conj (map (interiorizaNeg) (map (Neg) fs))
interiorizaNeg (Neg (Conj fs)) = Disy (map (interiorizaNeg) (map (Neg) fs))
interiorizaNeg (Neg (Neg f)) = interiorizaNeg f
interiorizaNeg (Neg f) = Neg (interiorizaNeg f)
interiorizaNeg (Disy fs) = Disy (map interiorizaNeg fs)
interiorizaNeg (Conj fs) = Conj (map interiorizaNeg fs)

```

Definimos (`interiorizaDisy f`) para interiorizar las disyunciones

```

interiorizaDisy :: Form -> Form
interiorizaDisy (Disy fs)
  | all (literal) fs = Disy fs
  | otherwise =

```

```

    Conj (map (aux2) (map (Disy) (concat (aux [ aux1 f | f <- fs]))))
    where
        aux [] = []
        aux (xs:xss) = map (combina xs) xss ++ aux xss
        aux1 p | literal p = [p]
        aux1 (Conj xs) = xs
        aux1 (Disy xs) = xs
        combina [] ys = []
        combina xs [] = []
        combina xs ys = [[x,y] | x <- xs, y <- ys]
        aux2 f | enFormaNC f = f
                | otherwise = interiorizaDisy f

interiorizaDisy f = f

```

*Nota 4.2.3.* Explicación de las funciones auxiliares

- La función `aux` aplica la función `combina` las listas de las conjunciones.
- La función `aux1` toma las listas de las conjunciones, construye una lista de un literal o unifica disyunciones.
- La función `combina xs ys` elabora listas de dos elementos de las listas `xs` e `ys`.
- La función `aux2` itera para interiorizar todas las disyunciones.

Debido a la representación que hemos elegido, pueden darse conjunciones de conjunciones, lo cual no nos interesa. Por ello, definimos `unificacionConjuncion` que extrae la conjunción al exterior.

```

unificaConjuncion :: Form -> Form
unificaConjuncion p@(Atom _ _) = p
unificaConjuncion (Disy fs) = Disy fs
unificaConjuncion (Conj fs) = Conj (concat (map (aux) (concat xs)))
    where
        xs = [ aux f | f <- fs]
        aux (Conj xs) = xs
        aux f = [f]

```

Por ejemplo,

```

-- | Ejemplos
-- >>> let f1 = Conj [p, Conj [r,q]]
-- >>> f1
-- (p^(r^q))
-- >>> unificaConjuncion f1
-- (p^(r^q))
-- >>> unificaConjuncion f1 == (Conj [p, Conj [r,q]])

```

```
-- False
-- >>> unificaConjuncion f1 == (Conj [p,r,q])
-- True
```

*Nota 4.2.4.* La representación “visual” por pantalla de una conjunción de conjunciones y su unificación puede ser la misma, como en el ejemplo anterior.

Así, hemos construido el algoritmo para el cálculo de formas normales conjuntivas. Definimos la función (`formaNormalConjuntiva f`)

```
formaNormalConjuntiva :: Form -> Form
formaNormalConjuntiva =
    unificaConjuncion . interiorizaDisy . interiorizaNeg . elimImpEquiv
```

Por ejemplo

```
-- | Ejemplos
-- >>> let f1 = Neg (Conj [p, Impl q r])
-- >>> f1
-- ¬(p ∧ (q ⇒ r))
-- >>> formaNormalConjuntiva f1
-- ((¬p ∨ q) ∧ (¬p ∨ ¬r))
-- >>> enFormaNC (formaNormalConjuntiva f1)
-- True
--
-- >>> let f2 = Neg (Conj [Disy [p,q],r])
-- >>> f2
-- ¬((p ∨ q) ∧ r)
-- >>> formaNormalConjuntiva f2
-- ((¬p ∨ ¬r) ∧ (¬q ∨ ¬r))
-- >>> enFormaNC (formaNormalConjuntiva f2)
-- True
--
-- >>> let f3 = (Impl (Conj [p,q]) (Disy [Conj [Disy [r,q],Neg p], Neg r]))
-- >>> f3
-- ((p ∧ q) ⇒ ((r ∨ q) ∧ ¬p) ∨ ¬r)
-- >>> formaNormalConjuntiva f3
-- ((¬p ∨ r) ∧ ((¬p ∨ q) ∧ ((¬p ∨ ¬p) ∧ ((¬p ∨ ¬r) ∧ ((¬q ∨ r) ∧ ((¬q ∨ q) ∧ ((¬q ∨ ¬p) ∧ (¬q ∨ ¬r)))))
-- >>> enFormaNC (formaNormalConjuntiva f3)
-- True
```

**Definición 4.2.2.** Una fórmula está en **forma normal disyuntiva** si es una disyunción de conjunciones de literales.

$$(p_1 \wedge \cdots \wedge p_n) \vee \cdots \vee (q_1 \wedge \cdots \wedge q_m)$$

### 4.2.2. Forma rectificada

**Definición 4.2.3.** Una fórmula  $F$  está en forma **rectificada** si ninguna variable aparece, de manera simultánea, libre y ligada, y cada cuantificador se refiere a una variable diferente.

Para proceder a su implementación definimos una función auxiliar previa que denotamos  $(\text{sustAux } n \ v \ f)$  que efectúa una sustitución de la variable  $v$  por  $x_n$ .

```
sustAux :: Int -> Variable -> Form -> Form
sustAux n v (PTodo var f)
  | var == v =
    PTodo (Variable "x" [n])
      (sustAux n v (sustitucionForm [(v, Var (Variable "x" [n]))] f))
  | otherwise = sustAux (n+1) var (PTodo var f)
sustAux n v (Ex var f)
  | var == v =
    Ex (Variable "x" [n])
      (sustAux n v (sustitucionForm [(v, Var (Variable "x" [n]))] f))
  | otherwise = sustAux (n+1) var (Ex var f)
sustAux n v (Impl f1 f2) =
  Impl (sustAux n v f1) (sustAux (n+k) v f2)
  where
    k = length (varEnForm f1)
sustAux n v (Conj fs) = Conj (map (sustAux n v) fs)
sustAux n v (Disy fs) = Disy (map (sustAux n v) fs)
sustAux n v (Neg f) = Neg (sustAux n v f)
sustAux n v f = sustitucionForm [(v, Var (Variable "x" [n]))] f
```

Añadimos ejemplos

```
-- | Ejemplo
-- >>> let f1 = PTodo x (Impl (Atom "P" [tx]) (Atom "Q" [tx,ty]))
-- >>> f1
--  $\forall x (P[x] \implies Q[x,y])$ 
-- >>> sustAux 0 x f1
--  $\forall x_0 (P[x_0] \implies Q[x_0,y])$ 
```

Definimos  $(\text{formaRectificada } f)$  que calcula la forma rectificada de la fórmula  $f$ .

```
formaRectificada :: Form -> Form
formaRectificada f@(PTodo x form) = sustAux 0 x f
formaRectificada f@(Ex x form) = sustAux 0 x f
formaRectificada (Impl f1 f2) =
  Impl (formaRectificada f1) (formaRectificada f2)
formaRectificada (Conj fs) = Conj (map formaRectificada fs)
formaRectificada (Disy fs) = Disy (map formaRectificada fs)
```

```
formaRectificada (Neg f) = Neg (formaRectificada f)
formaRectificada f = f
```

Por ejemplo

```
-- | Ejemplos
-- >>> formula2
--  $\forall x \forall y (R[x,y] \implies \exists z (R[x,z] \wedge R[z,y]))$ 
-- >>> formaRectificada formula2
--  $\forall x_0 \forall x_1 (R[x_0,x_1] \implies \exists x_4 (R[x_0,x_4] \wedge R[x_4,x_1]))$ 
-- >>> formula3
--  $(R[x,y] \implies \exists z (R[x,z] \wedge R[z,y]))$ 
-- >>> formaRectificada formula3
--  $(R[x,y] \implies \exists x_0 (R[x,x_0] \wedge R[x_0,y]))$ 
```

### 4.2.3. Forma normal prenexa

**Definición 4.2.4.** Una fórmula  $F$  está en forma **normal prenexa** si es de la forma  $Q_1x_1 \dots Q_nx_nG$  donde  $Q_i \in \{\forall, \exists\}$  y  $G$  no tiene cuantificadores.

Para la definición de la forma normal prenexa requerimos de dos funciones previas. Una que elimine los cuantificadores, (`eliminaCuant f`), y otra que los recolecta en una lista, (`recolectaCuant f`).

La definición de `eliminaCuant f` es

```
eliminaCuant :: Form -> Form
eliminaCuant (Ex x f) = eliminaCuant f
eliminaCuant (PTodo x f) = eliminaCuant f
eliminaCuant (Conj fs) = Conj (map eliminaCuant fs)
eliminaCuant (Disy fs) = Disy (map eliminaCuant fs)
eliminaCuant (Neg f) = Neg (eliminaCuant f)
eliminaCuant (Impl f1 f2) = Impl (eliminaCuant f1) (eliminaCuant f2)
eliminaCuant p@(Atom _ _) = p
```

Algunos ejemplos

```
-- | Ejemplos
-- >>> eliminaCuant formula2
--  $(R[x,y] \implies (R[x,z] \wedge R[z,y]))$ 
-- >>> eliminaCuant formula3
--  $(R[x,y] \implies (R[x,z] \wedge R[z,y]))$ 
```

La implementación de (`recolectaCuant f`) es

```

recolectaCuant :: Form -> [Form]
recolectaCuant (Ex x f) = (Ex x p): recolectaCuant f
recolectaCuant (PTodo x f) = (PTodo x p): recolectaCuant f
recolectaCuant (Conj fs) = concat (map recolectaCuant fs)
recolectaCuant (Disy fs) = concat (map recolectaCuant fs)
recolectaCuant (Neg f) = recolectaCuant f
recolectaCuant (Impl f1 f2) = recolectaCuant f1 ++ recolectaCuant f2
recolectaCuant p@(Atom _ _) = []

```

Por ejemplo,

```

-- | Ejemplos
-- >>> recolectaCuant formula2
-- [∀x p,∀y p,∃z p]
-- >>> recolectaCuant formula3
-- [∃z p]

```

Definimos la función `formaNormalPrenexa f` que calcula la forma normal prenexa de la fórmula `f`

```

formaNormalPrenexa :: Form -> Form
formaNormalPrenexa f = aplica cs (eliminaCuant (formaRectificada f))
  where
    aplica [] f = f
    aplica ((PTodo x _):fs) f = aplica fs (PTodo x f)
    aplica ((Ex x _):fs) f = aplica fs (Ex x f)
    cs = reverse (recolectaCuant (formaRectificada f))

```

Por ejemplo,

```

-- | Ejemplos
-- >>> formaNormalPrenexa formula2
-- ∀x0 ∀x1 ∃x4 (R[x0,x1]⇒(R[x0,x4]∧R[x4,x1]))
-- >>> formaNormalPrenexa formula3
-- ∃x0 (R[x,y]⇒(R[x,x0]∧R[x0,y]))

```

#### 4.2.4. Forma normal prenexa conjuntiva

**Definición 4.2.5.** Una fórmula  $F$  está en **forma normal prenexa conjuntiva** si está en forma normal prenexa con  $G$  en forma normal conjuntiva.

La implementamos en Haskell mediante la función `(formaNPConjuntiva f)`

```

formaNPConjuntiva :: Form -> Form
formaNPConjuntiva f = aux (formaNormalPrenexa f)
  where
    aux (PTodo x f) = PTodo x (aux f)
    aux (Ex x f) = Ex x (aux f)
    aux f = formaNormalConjuntiva f

```

Por ejemplo,

```

-- | Ejemplos
-- >>> formula2
--  $\forall x \forall y (R[x,y] \implies \exists z (R[x,z] \wedge R[z,y]))$ 
-- >>> formaNormalPrenexa formula2
--  $\forall x_0 \forall x_1 \exists x_4 (R[x_0,x_1] \implies (R[x_0,x_4] \wedge R[x_4,x_1]))$ 
-- >>> formula3
--  $(R[x,y] \implies \exists z (R[x,z] \wedge R[z,y]))$ 
-- >>> formaNormalPrenexa formula3
--  $\exists x_0 (R[x,y] \implies (R[x,x_0] \wedge R[x_0,y]))$ 

```

#### 4.2.5. Forma de Skolem

**Definición 4.2.6.** La fórmula  $F$  está en **forma de Skolem** si es de la forma  $\forall x_1 \dots \forall x_n G$ , donde  $n \geq 0$  y  $G$  no tiene cuantificadores.

Para transformar una fórmula en forma de Skolem emplearemos sustituciones y unificaciones. Además, necesitamos eliminar las equivalencias e implicaciones. Para ello definimos la equivalencia y equisatisfacibilidad entre fórmulas.

**Definición 4.2.7.** Las fórmulas  $F$  y  $G$  son **equivalentes** si para toda interpretación valen lo mismo.

**Definición 4.2.8.** Las fórmulas  $F$  y  $G$  son **equisatisfacibles** si se cumple que ambas son satisfacibles o ninguna lo es.

Finalmente, definamos una cadena de funciones, para finalizar con `(skolem f)` que transforma  $f$  a su forma de Skolem.

Se define la función `(skol k vs)` que convierte una lista de variables a un término de Skolem. Al calcular la forma de skolem de una fórmula, las variables cuantificadas son sustituidas por lo que denotamos **término de skolem** para obtener una fórmula libre. Los términos de skolem están compuestos por las siglas “sk” y un entero que lo identifique.

*Nota 4.2.5.* El término de skolem está expresado con la misma estructura que los términos funcionales.



```
skol :: Int -> [Variable] -> Termino
skol k vs = Ter ("sk" ++ show k) [Var x | x <- vs]
```

Por ejemplo,

```
| skol 1 [x] == sk1[x]
```

Definimos la función  $(skf\ f\ vs\ pol\ k)$ , donde

1.  $f$  es la fórmula que queremos convertir.
2.  $vs$  es la lista de los cuantificadores (son necesarios en la recursión).
3.  $pol$  es la polaridad, es de tipo `Bool`.
4.  $k$  es de tipo `Int` y sirve como identificador de la forma de Skolem.

**Definición 4.2.9.** La **Polaridad** cuantifica las apariciones de las variables cuantificadas de la siguiente forma:

- Una cantidad de apariciones impar de  $x$  en la subfórmula  $F$  de  $\exists xF$  indica que  $x$  tiene una polaridad negativa en la fórmula.
- Una cantidad de apariciones par de  $x$  en la subfórmula  $F$  de  $\forall xF$  indica que  $x$  tiene una polaridad positiva en la fórmula.

```
skf :: Form -> [Variable] -> Bool -> Int -> (Form,Int)
skf (Atom n ts) _ _ k =
  (Atom n ts,k)
skf (Conj fs) vs pol k =
  (Conj fs',j)
  where (fs',j) = skfs fs vs pol k
skf (Disy fs) vs pol k =
  (Disy fs', j)
  where (fs',j) = skfs fs vs pol k
skf (PTodo x f) vs True k =
  (PTodo x f',j)
  where vs' = insert x vs
        (f',j) = skf f vs' True k
skf (PTodo x f) vs False k =
  skf (sustitucionForm b f) vs False (k+1)
  where b = [(x,skol k vs)]
skf (Ex x f) vs True k =
  skf (sustitucionForm b f) vs True (k+1)
  where b = [(x,skol k vs)]
skf (Ex x f) vs False k =
  (Ex x f',j)
```

```

    where vs' = insert x vs
          (f',j) = skf f vs' False k
skf (Neg f) vs pol k =
    (Neg f',j)
    where (f',j) = skf f vs (not pol) k

```

donde la skolemización de una lista está definida por

```

skfs :: [Form] -> [Variable] -> Bool -> Int -> ([Form], Int)
skfs [] _ _ k = ([], k)
skfs (f:fs) vs pol k = (f':fs', j)
    where (f', j1) = skf f vs pol k
          (fs', j) = skfs fs vs pol j1

```

La skolemización de una fórmula sin equivalencias ni implicaciones se define por

```

sk :: Form -> Form
sk f = fst (skf f [] True 0)

```

La función (skolem f) devuelve la forma de Skolem de la fórmula f.

```

skolem :: Form -> Form
skolem = sk . elimImpEquiv

```

Por ejemplo,

```

-- | Ejemplos
-- >>> skolem formula2
--  $\forall x \forall y (\neg R[x,y] \vee (R[x, sk0[x,y]] \wedge R[sk0[x,y], y]))$ 
-- >>> skolem formula3
--  $(\neg R[x,y] \vee (R[x, sk0] \wedge R[sk0, y]))$ 
-- >>> skolem formula4
--  $R[cero, sk0]$ 
-- >>> skolem formula5
--  $(\neg P[sk0] \vee \forall y Q[x, y])$ 

```

### 4.3. Forma clausal

**Definición 4.3.1.** Un **literal** es un átomo o la negación de un átomo.

**Definición 4.3.2.** Una **cláusula** es un conjunto finito de literales.

**Definición 4.3.3.** Una **forma clausal** de una fórmula  $F$  es un conjunto de cláusulas equivalente a  $F$ .

*Proposición 4.3.1.* Si  $(p_1 \vee \dots \vee p_n) \wedge \dots \wedge (q_1 \vee \dots \vee q_m)$  es una forma normal conjuntiva de la fórmula  $F$ . Entonces, es una forma clausal de  $F$  es  $\{(p_1 \vee \dots \vee p_n), \dots, (q_1 \vee \dots \vee q_m)\}$

Por ejemplo una forma clausal de  $\neg(p \wedge (q \rightarrow r))$  es  $\{\{\neg p, q\}, \{\neg p, \neg r\}\}$

Se definen los tipos de dato `Clausula` y `Clausulas`, para representar una cláusula o un conjunto de ellas respectivamente.

```
data Clausula  = C [Form]
data Clausulas = Cs [Clausula]
```

Definimos su representación

```
instance Show Clausula where
  show (C []) = "[]"
  show (C fs) = "{" ++ init (tail (show fs)) ++ "}"
instance Show Clausulas where
  show (Cs []) = "[]"
  show (Cs cs) = "{" ++ init (tail (show cs)) ++ "}"
```

Por ejemplo de

```
-- | Fórmula
-- >>> Neg (Conj [p, Impl q r])
-- ¬(p ∧ (q ⇒ r))
```

su forma clausal es

```
-- | Forma clausal
-- >>> Cs [C [Neg p, q], C [Neg p, Neg r]]
-- {{¬p, q}, {¬p, ¬r}}
```

El algoritmo del cálculo de la forma clausal de una fórmula  $F$  es:

1. Sea  $F_1 = \exists y_1 \dots \exists y_n F$ , donde  $y_i$  con  $i = 1, \dots, n$  son las variables libres de  $F$ .
2. Sea  $F_2$  una forma normal prenexa conjuntiva rectificada de  $F_1$ .
3. Sea  $F_3 = \text{Skolem } (F_2)$ , que tiene la forma

$$\forall x_1 \dots \forall x_p [(L_1 \vee \dots \vee L_n) \wedge \dots \wedge (M_1 \vee \dots \vee M_m)]$$

Entonces, una forma clausal es

$$S = \{\{L_1, \dots, L_n\}, \dots, \{M_1, \dots, M_m\}\}$$

Dada una fórmula que está en la forma del paso 3 del algoritmo, es decir

$$f = \forall x_1 \dots \forall x_p [(L_1 \vee \dots \vee L_n) \wedge \dots \wedge (M_1 \vee \dots \vee M_m)]$$

, podemos convertirla a su forma clausal por medio de la función (`form3CAC f`)

```
form3CAC :: Form -> Clausulas
form3CAC (PTodo x f) = form3CAC f
form3CAC (Conj fs) = Cs (map disyAClau fs)
  where
    disyAClau p@(Atom _ _) = C [p]
    disyAClau (Disy fs) = C fs
```

Por ejemplo

```
-- | Ejemplo
-- >>> Conj [p, Disy [q,r]]
-- (p^(q^r))
-- >>> form3CAC (Conj [p, Disy [q,r]])
-- {{p},{q,r}}
```

Definimos (`formaClausal f`) que transforma `f` a su forma clausal.

```
formaClausal :: Form -> Clausulas
formaClausal = form3CAC . skolem . formaNPConjuntiva
```

Por ejemplo

```
-- | Ejemplo
-- >>> formaClausal (Neg (Conj [p, Impl q r]))
-- {{¬p,q},{¬p,¬r}}
```

## 4.4. Tableros semánticos

**Definición 4.4.1.** Un conjunto de fórmulas es **consistente** si tiene algún modelo. En caso contrario, se denomina **inconsistente**.

Distinguir el caso de fórmulas con variables libres.

La idea de obtener fórmulas equivalentes nos hace introducir los tipos de fórmulas alfa, beta, gamma y delta. No son más que equivalencias ordenadas por orden teórico en el que se pueden acometer para una simplificación eficiente de una fórmula, a otra cuyas únicas conectivas lógicas sean disyunciones y conjunciones.

■ Fórmulas alfa

$\neg(F_1 \rightarrow F_2)$	$F_1 \wedge F_2$
$\neg(F_1 \vee F_2)$	$F_1 \wedge \neg F_2$
$F_1 \leftrightarrow F_2$	$(F_1 \rightarrow F_2) \wedge (F_2 \rightarrow F_1)$

Las definimos en Haskell

```
alfa :: Form -> Bool
alfa (Conj _)      = True
alfa (Neg (Disy _)) = True
alfa _             = False
```

■ Fórmulas beta

$F_1 \rightarrow F_2$	$\neg F_1 \vee F_2$
$\neg(F_1 \wedge F_2)$	$\neg F_1 \vee \neg F_2$
$\neg(F_1 \leftrightarrow F_2)$	$\neg(F_1 \rightarrow F_2) \vee (\neg F_2 \rightarrow F_1)$

Las definimos en Haskell

```
beta :: Form -> Bool
beta (Disy _)      = True
beta (Neg (Conj _)) = True
beta _             = False
```

■ Fórmulas gamma

$\forall x F$	$F[x/t]$
$\neg \exists x F$	$\neg F[x/t]$

Notar que  $t$  es un término básico.

Las definimos en Haskell

```

gamma :: Form -> Bool
gamma (PTodo _ _)      = True
gamma (Neg (Ex _ _))   = True
gamma _                = False

```

#### ■ Fórmulas delta

$\exists xF$	$F[x/a]$
$\neg\forall F$	$\neg F[x/a]$

Notar que  $a$  es una constante nueva.

Las definimos en Haskell

```

delta :: Form -> Bool
delta (Neg (PTodo _ _)) = True
delta (Ex _ _)          = True
delta _                 = False

```

*Nota 4.4.1.* Cada elemento de la izquierda de las tablas es equivalente a la entrada de la derecha de la tabla que esté en su misma altura. Es decir, considerando las tablas como matrices  $a_{i,1} \equiv a_{i,2} \forall i$ .

Mediante estas equivalencias se procede a lo que se denomina método de los tableros semánticos. Uno de los objetivos del método de los tableros es determinar si una fórmula es consistente, así como la búsqueda de modelos.

**Definición 4.4.2.** Un **literal** es un átomo o la negación de un átomo.

Lo definimos en haskell

```

atomo, negAtomo, literal :: Form -> Bool

atomo (Atom _ _)      = True
atomo _               = False

negAtomo (Neg (Atom _ _)) = True
negAtomo _               = False

literal f = atomo f || negAtomo f

```

El método de tableros de un conjunto de fórmulas  $S$  sigue el siguiente algoritmo:


- El árbol cuyo único nodo tiene como etiqueta  $S$  es un tablero de  $S$ .
- Sea  $\mathcal{T}$  un tablero de  $S$  y  $S_1$  la etiqueta de una hoja de  $\mathcal{T}$ .
  1. Si  $S_1$  contiene una fórmula y su negación, entonces el árbol obtenido añadiendo como hijo de  $S_1$  el nodo etiquetado con  $\{\perp\}$  es un tablero de  $S$ .
  2. Si  $S_1$  contiene una doble negación  $\neg\neg F$ , entonces el árbol obtenido añadiendo como hijo de  $S_1$  el nodo etiquetado con  $(S_1 \setminus \{\neg\neg F\}) \cup \{F\}$  es un tablero de  $S$ .
  3. Si  $S_1$  contiene una fórmula alfa  $F$  de componentes  $F_1$  y  $F_2$ , entonces el árbol obtenido añadiendo como hijo de  $S_1$  el nodo etiquetado con  $(S_1 \setminus \{F\}) \cup \{F_1, F_2\}$  es un tablero de  $S$ .
  4. Si  $S_1$  contiene una fórmula beta de  $F$  de componentes  $F_1$  y  $F_2$ , entonces el árbol obtenido añadiendo como hijos de  $S_1$  los nodos etiquetados con  $(S_1 \setminus \{F\}) \cup \{F_1\}$  y  $(S_1 \setminus \{F\}) \cup \{F_2\}$  es un tablero de  $S$ .

**Definición 4.4.3.** Se dice que una hoja es **cerrada** si contiene una fórmula y su negación. Se representa  $\perp$

**Definición 4.4.4.** Se dice que una hoja es **abierta** si es un conjunto de literales y no contiene un literal y su negación.

**Definición 4.4.5.** Un **tablero completo** es un tablero tal que todas sus hojas son abiertas o cerradas.

Ejemplo de tablero completo



Se solapan las ramas del árbol

Representamos la fórmula de este tablero en Haskell.

Representamos la fórmula

```


tab1 = Neg (Impl (Disy [p,q]) (Conj [p,q]))

-- | Representación
-- >>> tab1
-- ¬((p∨q)⇒(p∧q))

```

**Definición 4.4.6.** Un tablero es **cerrado** si todas sus hojas son cerradas.

Un ejemplo de tablero cerrado es



La fórmula del tablero se representa en Haskell

```

tab2 = Conj [Impl p q, Impl q r, Neg (Impl p r)]
-- | Representación
-- >>> tab2
-- ((p⇒q)∧((q⇒r)∧¬(p⇒r)))

```

**Teorema 4.4.1.** Si una fórmula  $F$  es consistente, entonces cualquier tablero de  $F$  tendrá ramas abiertas.

Nuestro objetivo es definir en Haskell un método para el cálculo de tableros semánticos. El contenido relativo a tableros semánticos se encuentra en el módulo Tableros.

```

module Tableros where
import DNH
import PTL
import LPH
import Debug.Trace

```



Hemos importado la librería `Debug.Trace` porque emplearemos la función `trace`. Esta función tiene como argumentos una cadena de caracteres, una función, y un valor sobre el que se aplica la función. Por ejemplo

```
ghci> trace ("aplicando even a x = " ++ show 3) (even 3)
aplicando even a x = 3
False
```

A lo largo de esta sección trabajaremos con fórmulas en su forma de Skolem.

El método de tableros se suele representar en forma de árbol, por ello definiremos el tipo de dato `Nodo`.

```
data Nodo = Nd Indice [Termino] [Termino] [Form]
           deriving Show
```

Donde la primera lista de términos representa los literales positivos, la segunda lista de términos representa los negativos, y la lista de fórmulas son aquellas ligadas a los términos de las listas anteriores.

Definimos los tableros como una lista de nodos.

```
type Tablero = [Nodo]
```

Necesitamos poder reconocer las dobles negaciones, para ello definimos la función `dobleNeg f`.

```
dobleNeg (Neg (Neg f)) = True
dobleNeg _             = False
```

Una función auxiliar de conversión de literales a términos es `literalATer t`.

```
literalATer :: Form -> Termino
literalATer (Atom n ts)      = Ter n ts
literalATer (Neg (Atom n ts)) = Ter n ts
```

Definimos la función `(componentes f)` que determina los componentes de una fórmula `f`.

```
componentes :: Form -> [Form]
componentes (Conj fs)      = fs
componentes (Disy fs)      = fs
componentes (Neg (Conj fs)) = [Neg f | f <- fs]
componentes (Neg (Disy fs)) = [Neg f | f <- fs]
componentes (Neg (Neg f))  = [f]
componentes (PTodo x f)    = [f]
componentes (Neg (Ex x f))  = [Neg f]
```

Por ejemplo,

```
ghci> componentes (skolem (tab1))
[¬¬(p∧[q]),¬(p∨[q])]
ghci> componentes (skolem (tab2))
[(¬p∨[q]),(¬q∨[r]),¬(¬p∨[r])]
```

Definimos la función (`varLigada f`) que devuelve la variable ligada de la fórmula `f`

```
varLigada :: Form -> Variable
varLigada (PTodo x f)    = x
varLigada (Neg (Ex x f)) = x
```

Definimos la función (`descomponer f`) que determina los cuantificadores universales de `f`.

```
descomponer :: Form -> ([Variable],Form)
descomponer = descomp [] where
  descomp xs f | gamma f    = descomp (xs ++ [x]) g
                | otherwise = (xs,f)
    where x    = varLigada f
          [g] = componentes f
```

Por ejemplo,

```
ghci> formula2
∀x ∀y (R[x,y]⇒∃z (R[x,z]∧R[z,y]))
ghci> descomponer formula2
([x,y],(R[x,y]⇒∃z (R[x,z]∧R[z,y])))
ghci> formula3
(R[x,y]⇒∃z (R[x,z]∧R[z,y]))
ghci> descomponer formula3
([],(R[x,y]⇒∃z (R[x,z]∧R[z,y])))
ghci> formula4
∃x R[cero,x]
ghci> descomponer formula4
([],∃x R[cero,x])
```

Definimos (`ramificacion nodo`) que ramifica un nodo aplicando las equivalencias adecuadas.

```
ramificacion :: Nodo -> Tablero
ramificacion (Nd i pos neg []) = [Nd i pos neg []]
ramificacion (Nd i pos neg (f:fs))
  | atomo f    = if literalATer f 'elem' neg
                  then []
```

```

        else [Nd i ( literalATer f : pos) neg fs]
| negAtomo f = if literalATer f 'elem' pos
    then []
    else [Nd i pos (literalATer f : neg) fs]
| dobleNeg f = [Nd i pos neg (componentes f ++ fs)]
| alfa f      = [Nd i pos neg (componentes f ++ fs)]
| beta f      = [Nd (i++[n]) pos neg (f':fs)
    | (f',n) <- zip (componentes f) [0..]]
| gamma f     = [Nd i pos neg (f':(fs++[f]))]
where
  (xs,g) = descomponer f
  b      = [(Variable nombre j, Var (Variable nombre i))
    | (Variable nombre j) <- xs]
  f'      = sustitucionForm b g

```

Debido a que pueden darse la infinitud de un árbol por las fórmulas gamma, definimos otra función (ramificacionP k nodo) que ramifica un nodo teniendo en cuenta la profundidad.

```

ramificacionP :: Int -> Nodo -> (Int,Tablero)
ramificacionP k nodo@(Nd i pos neg []) = (k,[nodo])
ramificacionP k (Nd i pos neg (f:fs))
  | atomo    f = if literalATer f 'elem' neg
    then (k,[])
    else (k,[Nd i (literalATer f : pos) neg fs])
  | negAtomo f = if literalATer f 'elem' neg
    then (k,[])
    else (k,[Nd i pos (literalATer f : neg) fs])
  | dobleNeg f = (k,[Nd i pos neg (componentes f ++ fs)])
  | alfa      f = (k,[Nd i pos neg (componentes f ++ fs)])
  | beta      f = (k,[Nd (i++[n]) pos neg (f':fs)
    | (f',n) <- zip (componentes f) [0..] ])
  | gamma     f = (k-1, [Nd i pos neg (f':(fs++[f]))])
where
  (xs,g) = descomponer f
  b      = [(Variable nombre j, Var (Variable nombre i))
    | (Variable nombre j) <- xs]
  f'      = sustitucionForm b g

```

**Definición 4.4.7.** Un nodo está completamente **expandido** si no se puede seguir ramificando

Se define en Haskell

```

nodoExpandido :: Nodo -> Bool
nodoExpandido (Nd i pos neg []) = True
nodoExpandido _                  = False

```

Definimos la función (`expandeTablero n tab`) que desarrolla un tablero a una profundidad `n`.

```
expandeTablero :: Int -> Tablero -> Tablero
expandeTablero 0 tab = tab
expandeTablero _ [] = []
expandeTablero n (nodo:nodos)
  | nodoExpandido nodo = nodo : expandeTablero n nodos
  | k == n              = expandeTablero n (nuevoNodo ++ nodos)
  | otherwise           = expandeTablero (n-1) (nodos ++ nuevoNodo)
  where (k,nuevoNodo) = ramificacionP n nodo
```

Para una visualización más gráfica, definimos (`expandeTableroG`) empleando la función (`trace`).

```
expandeTableroG :: Int -> Tablero -> Tablero
expandeTableroG 0 tab = tab
expandeTableroG _ [] = []
expandeTableroG n (nodo:nodos)
  | nodoExpandido nodo =
      trace (show nodo ++ "\n\n") (nodo : expandeTableroG n nodos)
  | k == n =
      trace (show nodo ++ "\n\n") (expandeTableroG k
                                   (nuevoNodo ++ nodos))
  | otherwise =
      trace (show nodo ++ "\n\n") (expandeTableroG (n-1)
                                   (nodos ++ nuevoNodo))
  where (k, nuevoNodo) = ramificacionP n nodo
```

Definimos la función (`esNodoCerrado`) para comprobar si hay hoja cerrada.

```
esNodoCerrado :: Nodo -> [Sust]
esNodoCerrado (Nd _ pos neg _) =
  concat [ unificadoresTerminos p n | p <- pos,
                                       n <- neg ]
```

Definimos las funciones auxiliares (`sustNodo nd`) y (`sustTab tb`) que aplican sustituciones a nodos y tableros.

```
sustNodo :: Sust -> Nodo -> Nodo
sustNodo b (Nd i pos neg f) =
  Nd i (susTerms b pos) (susTerms b neg) (sustitucionForms b f)

susTab :: Sust -> Tablero -> Tablero
susTab = map . sustNodo
```

Se define `esCerrado` para determinar si un tablero es cerrado.

```
esCerrado :: Tablero -> [Sust]
esCerrado []      = [identidad]
esCerrado [nodo] = esNodoCerrado nodo
esCerrado (nodo:nodos) =
  concat [esCerrado (susTab s nodos) | s <- esNodoCerrado nodo ]
```

Dada una fórmula es necesario crear un tablero inicial para posteriormente desarrollarlo. Lo hacemos mediante la función `(tableroInicial f)`.

```
tableroInicial :: Form -> Tablero
tableroInicial f = [Nd [] [] [] [f]]
```

Por ejemplo,

```
ghci> tab1
¬((p∨q)⇒(p∧q))
ghci> tableroInicial tab1
[Nd [] [] [] [¬((p∨[q])⇒(p∧[q]))]]
```

La función `(refuta k f)` intenta refutar la fórmula `f` con un tablero de profundidad `k`.

```
refuta :: Int -> Form -> Bool
refuta k f = esCerrado tab /= []
  where tab = expandeTablero k (tableroInicial f)
```

*Nota 4.4.2.* Se puede emplear también `expandeTableroG`, por ello se deja comentado para su posible uso.

**Definición 4.4.8.** Una fórmula  $F$  es un **teorema** mediante tableros semánticos si tiene una prueba mediante tableros; es decir, si  $\neg F$  tiene un tablero completo cerrado

Finalmente, podemos determinar si una fórmula es un teorema y si es satisfacible mediante las funciones `(esTeorema n f)` y `(satisfacible n f)`.

```
esTeorema, satisfacible :: Int -> Form -> Bool
esTeorema n = refuta n . skolem . Neg
satisfacible n = not . refuta n . skolem
```

Por ejemplo tomando `tautologia1` y la ya usada anteriormente `formula2`

```
tautologia1 :: Form
tautologia1 = Disy [Atom "P" [tx], Neg (Atom "P" [tx])]
```

se tiene

```
ghci> tautologia1
(P[x] ∨ ¬P[x])
ghci> esTeorema 1 tautologia1
True
ghci> formula2
∀x ∀y (R[x,y] ⇒ ∃z (R[x,z] ∧ R[z,y]))
ghci> esTeorema 20 formula2
False
ghci> tab1
¬((p ∨ q) ⇒ (p ∧ q))
ghci> esTeorema 20 tab1
False
ghci> satisfacible 1 tab1
True
ghci> tab2
((p ⇒ q) ∧ ((q ⇒ r) ∧ ¬(p ⇒ r)))
ghci> esTeorema 20 tab2
False
ghci> satisfacible 20 tab2
False
```

**Teorema 4.4.2.** *El cálculo de tableros semánticos es adecuado y completo.*

**Definición 4.4.9.** Una fórmula  $F$  es **deducible** a partir del conjunto de fórmulas  $S$  si existe un tablero completo cerrado de la conjunción de  $S$  y  $\neg F$ . Se representa por  $S \vdash_{Tab} F$ .

Explicar más el método de tableros con polaridad.

Comparar la implementación con la de Ben Ari que se encuentra en <https://github.com/motib/mlcs/blob/master/fol/tab1.pro>

# Capítulo 5

## Modelos de Herbrand

En este capítulo se pretende formalizar los modelos de Herbrand. Herbrand propuso un método constructivo para generar interpretaciones de fórmulas o conjuntos de fórmulas.

El contenido de este capítulo se encuentra en el módulo Herbrand.

```
module Herbrand where
import Data.List
import Text.PrettyPrint.GenericPretty (pp)
import PFH
import LPH
import PTLP
```

### 5.1. Universo de Herbrand

**Definición 5.1.1.** El **universo de Herbrand** de  $L$  es el conjunto de términos básicos de  $F$ . Se representa por  $\mathcal{UH}(L)$ .

*Proposición 5.1.1.*  $\mathcal{UH}(L) = \bigcup_{i \geq 0} H_i(L)$ , donde  $H_i(L)$  es el nivel  $i$  del  $\mathcal{UH}(L)$  definido por

$$H_0(L) = \begin{cases} C, & \text{si } C \neq \emptyset \\ a, & \text{en caso contrario (a nueva constante)} \end{cases}$$
$$H_{i+1}(L) = H_i(L) \cup \{f(t_1, \dots, t_n) : f \in \mathcal{F}_n \text{ y } t_i \in H_i(L)\}$$

A continuación caracterizamos las constantes. Definimos la función (`esConstante t`) que se verifica si el término  $t$  es una constante.

```
-- | Ejemplos
-- >>> esConstante a
-- True
```

```
-- >>> esConstante tx
-- False
esConstante :: Termino -> Bool
esConstante (Ter _ []) = True
esConstante _          = False
```

El valor de (`constantesTerm t`) es el conjunto de las constantes del término `t`.

```
-- | Ejemplos
-- >>> let t = Ter "f" [Ter "a" [], Ter "b" [], Ter "g" [tx, Ter "a" []]]
-- >>> t
-- f[a,b,g[x,a]]
-- >>> constantesTerm t
-- [a,b]
constantesTerm :: Termino -> [Termino]
constantesTerm (Var _)      = []
constantesTerm c@(Ter _ []) = [c]
constantesTerm (Ter f ts)   = nub (concatMap constantesTerm ts)
```

El valor (`constantesForm f`) es el conjunto de las constantes de la fórmula `f`.

```
-- | Ejemplos
-- >>> let f1 = Atom "R" [a,tx]
-- >>> f1
-- R[a,x]
-- >>> constantesForm f1
-- [a]
-- >>> let f2 = Conj [Atom "P" [a, Ter "f" [tx,b]], f1]
-- >>> f2
-- (P[a,f[x,b]]^R[a,x])
-- >>> constantesForm f2
-- [a,b]
-- >>> let f3 = PTodo x f2
-- >>> f3
-- ∀x (P[a,f[x,b]]^R[a,x])
-- >>> constantesForm f3
-- [a,b]
constantesForm :: Form -> [Termino]
constantesForm (Atom _ ts)   = nub (concatMap constantesTerm ts)
constantesForm (Neg f)       = constantesForm f
constantesForm (Impl f1 f2)  = constantesForm f1 'union' constantesForm f2
constantesForm (Equiv f1 f2) = constantesForm f1 'union' constantesForm f2
constantesForm (Conj fs)     = nub (concatMap constantesForm fs)
constantesForm (Disy fs)     = nub (concatMap constantesForm fs)
constantesForm (PTodo _ f)   = constantesForm f
constantesForm (Ex _ f)      = constantesForm f
```



La propiedad (`esFuncion t`) se verifica si  $t$  es un término compuesto (es decir, es un término pero no es una variable ni una constante).

```
-- | Ejemplos
-- >>> esFuncion (Ter "f" [a])
-- True
-- >>> esFuncion (Ter "f" [])
-- False
-- >>> esFuncion tx
-- False
esFuncion :: Termino -> Bool
esFuncion (Ter _ xs) = not (null xs)
esFuncion _          = False
```

(`funForm f`) para obtener todos los símbolos funcionales que aparezcan en la fórmula  $f$ .

```
funForm :: Form -> [Termino]
funForm (Atom _ ts) = nub [ t | t <- ts, esFuncion t ]
funForm (Neg f)     = funForm f
funForm (Impl f1 f2) = funForm f1 'union' funForm f2
funForm (Equiv f1 f2) = funForm f1 'union' funForm f2
funForm (Conj fs)    = nub (concatMap funForm fs)
funForm (Disy fs)    = nub (concatMap funForm fs)
funForm (PTodo x f)  = funForm f
funForm (Ex x f)     = funForm f
```

Por ejemplo

```
|ghci> funForm (Conj [Atom "P" [a, Ter "f" [tx,b]], Atom "R" [Ter "g" [tx,ty],c]])
|[f[x,b],g[x,y]]
```

**Definición 5.1.2.** La **aridad** de un operador  $f(x_1, \dots, x_n)$  es el número de argumentos a los que se aplica.

Definimos (`aridadF t`) para calcular la aridad del término  $t$ .

```
aridadF :: Termino -> Int
aridadF (Ter _ ts) = length ts
```

También necesitamos definir dos funciones auxiliares que apliquen los símbolos funcionales a las constantes del universo de Herbrand. Las funciones son (`aplicaFunAConst f c`), que aplica el símbolo funcional  $f$  a la constante  $c$  y (`aplicaFun fs cs`) que es una generalización a listas de la anterior.

```

aplicaFunAConst :: Termino -> [Termino] -> Termino
aplicaFunAConst (Ter s _) = Ter s

aplicaFun :: [Termino] -> [Termino] -> [Termino]
aplicaFun [] cs = []
aplicaFun (f:fs) cs =
    map (aplicaFunAConst f) (variacionesR (aridadF f) cs)
    ++ aplicaFun fs cs

```

Así podemos obtener el universo de Herbrand de una fórmula  $f$  definiendo (`univHerbrand n f`)

```

univHerbrand :: (Eq a, Num a) => a -> Form -> Universo Termino
univHerbrand 0 f = if constantesForm f /= [] then sort(constantesForm f)
                  else [a]
univHerbrand n f =
    sort (nub (univHerbrand (n-1) f ++ aplicaFun (funForm f)
                (univHerbrand (n-1) f)))

```

Por ejemplo

```

ghci> formula2
∀x ∀y (R[x,y]⇒∃z (R[x,z]∧R[z,y]))
ghci> univHerbrand 0 formula2
[a]

ghci> Conj [Disy [Atom "P" [a],Atom "P" [b]],
            Disy [Neg (Atom "P" [b]),Atom "P" [c]],
            Impl (Atom "P" [a]) (Atom "P" [c])]
((P[a]∨P[b])∧((¬P[b]∨P[c])∧(P[a]⇒P[c])))
ghci> univHerbrand 0 (Conj [Disy [Atom "P" [a],Atom "P" [b]],
                            Disy [Neg (Atom "P" [b]),Atom "P" [c]],
                            Impl (Atom "P" [a]) (Atom "P" [c])])
[a,b,c]

ghci> Conj [PTodo x (PTodo y (Impl (Atom "Q" [b,tx])
                                   (Disy [Atom "P" [a],Atom "R" [ty]]))),
            Impl (Atom "P" [b]) (Neg (Ex z (Ex u (Atom "Q" [tz,tu]))))]
(∀x ∀y (Q[b,x]⇒(P[a]∨R[y]))∧(P[b]⇒¬∃z ∃u Q[z,u]))
ghci> univHerbrand 0 (Conj [PTodo x (PTodo y (Impl (Atom "Q" [b,tx])
                                   (Disy [Atom "P" [a],Atom "R" [ty]]))),
                            Impl (Atom "P" [b]) (Neg (Ex z (Ex u (Atom "Q" [tz,tu]))))]
[a,b]

```

*Proposición 5.1.2.*  $\mathcal{UH}$  es finito si y sólo si no tiene símbolos de función.

Definimos fórmulas con términos funcionales para el ejemplo

```
formula6, formula7 :: Form
formula6 = PTodo x (Atom "P" [Ter "f" [tx]])
formula7 = PTodo x (Atom "P" [Ter "f" [a,b]])
```

quedando como ejemplo

```
ghci> formula6
∀x P[f[x]]
ghci> univHerbrand 5 formula6
[a,f[a],f[f[a]],f[f[f[a]]],f[f[f[f[a]]]],f[f[f[f[f[a]]]]]]
ghci> univHerbrand 0 formula7
[a,b]
ghci> univHerbrand 1 formula7
[a,b,f[a,a],f[a,b],f[b,a],f[b,b]]
ghci> univHerbrand 2 formula7
[a,b,f[a,a],f[a,b],f[a,f[a,a]],f[a,f[a,b]],f[a,f[b,a]],
f[a,f[b,b]],f[b,a],f[b,b],f[b,f[a,a]],f[b,f[a,b]],f[b,f[b,a]],
f[b,f[b,b]],f[f[a,a],a],f[f[a,a],b],f[f[a,a],f[a,a]],f[f[a,a],
f[a,b]],f[f[a,a],f[b,a]],f[f[a,a],f[b,b]],f[f[a,b],a],f[f[a,b],b],
f[f[a,b],f[a,a]],f[f[a,b],f[a,b]],f[f[a,b],f[b,a]],f[f[a,b],
f[b,b]],f[f[b,a],a],f[f[b,a],b],f[f[b,a],f[a,a]],f[f[b,a],f[a,b]],
f[f[b,a],f[b,a]],f[f[b,a],f[b,b]],f[f[b,b],a],f[f[b,b],b],f[f[b,b],
f[a,a]],f[f[b,b],f[a,b]],f[f[b,b],f[b,a]],f[f[b,b],f[b,b]]]
```

Hay que tener en cuenta que se dispara la cantidad de elementos del universo de Herbrand ante términos funcionales con aridad grande.

```
length (univHerbrand 0 formula7) == 2
length (univHerbrand 1 formula7) == 6
length (univHerbrand 2 formula7) == 38
length (univHerbrand 3 formula7) == 1446
```

## 5.2. Base de Herbrand

**Definición 5.2.1.** Una **fórmula básica** es una fórmula sin variables ni cuantificadores.

**Definición 5.2.2.** La **base de Herbrand**  $\mathcal{BH}(L)$  de un lenguaje  $L$  es el conjunto de átomos básicos de  $L$ .

Con el objetivo de definir una función que obtenga la base de Herbrand; necesitamos poder calcular el conjunto de símbolos de predicado de una fórmula.

Definimos  $(aridadP\ f)$  que determina la aridad del predicado de la fórmula atómica  $f$ .

```
aridadP :: Form -> Int
aridadP (Atom _ xs) = length xs
```

Por ejemplo para  $R(x, y, a)$  la aridad es 3

```
ghci> aridadP (Atom "R" [tx,ty,a])
3
```

Definimos (esPredicado f) que determina si f es un predicado.

```
esPredicado :: Form -> Bool
esPredicado (Atom _ []) = False
esPredicado (Atom _ _)  = True
esPredicado _            = False
```

Calculamos el conjunto de los predicados de una fórmula f definiendo la función (predicadosForm f).

```
predicadosForm :: Form -> [Form]
predicadosForm p@(Atom _ _) = [p | esPredicado p]
predicadosForm (Neg f)      = predicadosForm f
predicadosForm (Impl f1 f2) =
  predicadosForm f1 'union' predicadosForm f2
predicadosForm (Equiv f1 f2) =
  predicadosForm f1 'union' predicadosForm f2
predicadosForm (Conj fs)    = nub (concatMap predicadosForm fs)
predicadosForm (Disy fs)    = nub (concatMap predicadosForm fs)
predicadosForm (PTodo x f)  = predicadosForm f
predicadosForm (Ex x f)     = predicadosForm f
```

Esta función repite el mismo predicado si tiene distintos argumentos, como por ejemplo

```
ghci> formula2
∀x ∀y (R[x,y] ⇒ ∃z (R[x,z] ∧ R[z,y]))
ghci> predicadosForm formula2
[R[x,y], R[x,z], R[z,y]]
```

Por lo tanto, definimos (predForm f) que obtiene los símbolos de predicado sin repeticiones.

```
predForm :: Form -> [Form]
predForm = noRep . predicadosForm
  where
    noRep [] = []
    noRep (Atom st t : ps) =
      if null [Atom str ts | (Atom str ts) <- ps, str == st]
      then Atom st t : noRep ps else noRep ps
```

Así obtenemos

```
ghci> predForm formula2
[R[z,y]]
```

Podemos también obtener una lista de los símbolos de predicado definiendo (`simbolosPred f`)

```
simbolosPred :: Form -> [Nombre]
simbolosPred f = [str | (Atom str _) <- ps]
  where ps = predForm f
```

Definir directamente `simbolosPred` sin usar `predForm` y eliminar `predForm` y `predicadosForm`. Respuesta a comentario: La razón de no haber definido `simbolosPred` sin `predForm` y `predicadosForm` es por la necesidad de mantener la aridad del predicado para la definición de base de Herbrand.

Finalmente, necesitamos aplicar los símbolos de predicado al universo de Herbrand correspondiente.

Definimos las funciones (`aplicaPred p ts`) y su generalización (`apPred ps ts`) para aplicar los símbolos de predicado.

```
aplicaPred :: Form -> [Termino] -> Form
aplicaPred (Atom str _) = Atom str

apPred :: [Form] -> [Termino] -> [Form]
apPred [] ts = []
apPred (p:ps) ts = map (aplicaPred p) (variacionesR (aridadP p) ts)
  ++ apPred ps ts
```

Algunos ejemplos son

```
ghci> aplicaPred (Atom "P" [tx]) [ty]
P[y]
ghci> aplicaPred (Atom "R" [tx,ty]) [tz,ty]
R[z,y]
ghci> apPred [Atom "P" [tx]] [tx,ty,tz]
[P[z],P[y],P[x]]
ghci> apPred [Atom "P" [tx], Atom "R" [tx,ty]] [tx,ty,tz]
[P[z],P[y],P[x],R[y,z],R[z,y],R[x,z],R[z,x],R[x,y],R[y,x]]
```

Definimos la función (`baseHerbrand n f`)

```
baseHerbrand :: (Eq a, Num a) => a -> Form -> [Form]
baseHerbrand n f = nub (apPred (predForm f) (univHerbrand n f))
```

Algunos ejemplos

```
ghci> baseHerbrand 0 (Conj [Disy [Atom "P" [a],Atom "P" [b]],
                             Disy [Neg (Atom "P" [b]),Atom "P" [c]],
                             Impl (Atom "P" [a]) (Atom "P" [c])])
[P[a],P[b],P[c]]

ghci> baseHerbrand 0 (Conj [PTodo x (PTodo y (Impl (Atom "Q" [b,tx])
                                                    (Disy [Atom "P" [a],Atom "R" [ty]]))),
                             Impl (Atom "P" [b]) (Neg (Ex z (Ex u (Atom "Q" [tz,tu])))))]
[R[a],R[b],P[a],P[b],Q[a,a],Q[a,b],Q[b,a],Q[b,b]]
```

Podemos hacer un análisis de la fórmula 6, calculando sus constantes, símbolos funcionales y símbolos de predicado. Así como el universo de Herbrand y la base de Herbrand.

```
ghci> formula6
∀x P[f[x]]
ghci> constantesForm formula6
[]
ghci> funForm formula6
[f[x]]
ghci> simbolosPred formula6
["P"]
ghci> univHerbrand 0 formula6
[a]
ghci> univHerbrand 1 formula6
[a,f[a]]
ghci> univHerbrand 2 formula6
[a,f[a],f[f[a]]]
ghci> baseHerbrand 0 formula6
[P[a]]
ghci> baseHerbrand 1 formula6
[P[a],P[f[a]]]
ghci> baseHerbrand 2 formula6
[P[a],P[f[a]],P[f[f[a]]]]
```

**Teorema 5.2.1.**  $\mathcal{BH}(L)$  es finita si y sólo si  $L$  no tiene símbolos de función.

### 5.3. Interpretaciones de Herbrand

**Definición 5.3.1.** Una **interpretación de Herbrand** es una interpretación  $\mathcal{I} = (\mathcal{U}, I)$  tal que

- $\mathcal{U}$  es el universo de Herbrand de  $F$ .
- $I(c) = c$ , para constante  $c$  de  $F$ .
- $I(f) = f$ , para cada símbolo funcional de  $F$ .

## 5.4. Modelos de Herbrand

**Definición 5.4.1.** Un **modelo de Herbrand** de una fórmula  $F$  es una interpretación de Herbrand de  $F$  que es modelo de  $F$ .

Un **modelo de Herbrand** de un conjunto de fórmulas  $S$  es una interpretación de Herbrand de  $S$  que es modelo de  $S$ .

*Proposición 5.4.1.* Una interpretación de Herbrand queda determinada por un subconjunto de la base de Herbrand.

Definimos (`valHerbrand f n`) que determina si existe algún subconjunto de la base de Herbrand que sea modelo de la fórmula  $f$ . Para definirla necesitamos una función previa (`valorHerbrand f f n`) que realiza una recursión sobre la fórmula  $f$ , comprobando que exista algún elemento de la base de Herbrand que sea modelo de la fórmula. Finalmente, `valHerbrand` será una evaluación de `valorHerbrand`.

```

valorHerbrand :: Form -> Form -> Int -> Bool
valorHerbrand p@(Atom str ts) f n =
  p 'elem' baseHerbrand n f
valorHerbrand (Neg g) f n =
  not (valorHerbrand g f n)
valorHerbrand (Impl f1 f2) f n =
  valorHerbrand f1 f n <= valorHerbrand f2 f n
valorHerbrand (Equiv f1 f2) f n =
  valorHerbrand f1 f n == valorHerbrand f2 f n
valorHerbrand (Conj fs) f n =
  all (==True) [valorHerbrand g f n | g <- fs]
valorHerbrand (Disy fs) f n =
  True 'elem' [valorHerbrand g f n | g <- fs]
valorHerbrand (PTodo v g) f n =
  valorHerbrand g f n
valorHerbrand (Ex v g) f n =
  valorHerbrand g f n

```

*Nota 5.4.1.* Se puede cambiar la  $n$  de la base de Herbrand a la que se calcula la existencia de modelo. Eso es interesante para fórmulas con símbolos funcionales.

```

valHerbrand :: Form -> Int -> Bool
valHerbrand g = valorHerbrand f f
  where f = skolem g

```

Añadimos un par de fórmulas para los ejemplos

```

formula8 = Impl (Atom "P" [tx]) (Atom "Q" [tx])
formula9 = Conj [Atom "P" [tx], Neg (Atom "P" [tx])]

```

```
ghci> valHerbrand formula8 0
True
ghci> valHerbrand formula9 0
False
ghci> valHerbrand formula6 0
False
ghci> valHerbrand formula6 1
True
ghci> valHerbrand formula2 0
True
```

La fórmula 9 es una contradicción, es decir, no tiene ningún modelo. Podemos comprobarlo mediante tableros

```
ghci> satisfacible 1 formula9
False
```

## 5.5. Consistencia mediante modelos de Herbrand

*Proposición 5.5.1.* Sea  $S$  un conjunto de fórmulas básicas. Son equivalentes:

1.  $S$  es consistente.
2.  $S$  tiene un modelo de Herbrand.

*Proposición 5.5.2.* Existen conjuntos de fórmulas consistentes sin modelos de Herbrand.

Un ejemplo de fórmula consistente sin modelo de Herbrand

```
formula10 :: Form
formula10 = Conj [Ex x (Atom "P" [tx]), Neg (Atom "P" [a])]
```

Como podemos ver aplicando valHerbrand

```
ghci> formula10
(∃x P[x] ∧ ¬P[a])
ghci> valHerbrand formula10 0
False
ghci> valHerbrand formula10 1
False
ghci> valHerbrand formula10 2
False
ghci> valHerbrand formula10 3
False
```

Pero es satisfacible

```
ghci> satisfacible 0 formula10
True
```



## 5.6. Extensiones de Herbrand

**Definición 5.6.1.** Sea  $C = \{L_1, \dots, L_n\}$  una cláusula de  $L$  y  $\sigma$  una sustitución de  $L$ . Entonces,  $C\sigma = \{L_1\sigma, \dots, L_n\sigma\}$  es una **instancia** de  $C$ .

**Definición 5.6.2.**  $C\sigma$  es una **instancia básica** de  $C$  si todos los literales de  $C\sigma$  son básicos.

Por ejemplo, si tenemos  $C = \{P(x, a), \neg P(x, f(y))\}$ , una instancia básica sería

$$C[x/a, y/f(a)] = \{P(a, a), \neg P(x, f(f(a)))\}$$

Que en haskell lo habríamos representado por

```
ghci> Conj [Atom "P" [tx,a], Neg (Atom "P" [tx, Ter "f" [ty]])]
(P[x,a] ∧ ¬P[x,f[y]])
ghci> sustitucionForm [(x,a), (y, Ter "f" [a])]
      (Conj [Atom "P" [tx,a], Neg (Atom "P" [tx, Ter "f" [ty]])])
(P[a,a] ∧ ¬P[a,f[f[a]]])
```

**Definición 5.6.3.** La **extensión de Herbrand** de un conjunto de cláusulas  $Cs$  es el conjunto de fórmulas

$$EH(Cs) = \{C\sigma : C \in Cs \text{ y } \forall x \in C, \sigma(x) \in UH(Cs)\}$$

*Proposición 5.6.1.*  $EH(L) = \cup_{i \geq 0} EH_i(L)$ , donde  $EH_i(L)$  es el nivel  $i$  de la  $EH(L)$

$$EH_i(Cs) = \{C\sigma : C \in Cs \text{ y } \forall x \in C, \sigma(x) \in UH_i(Cs)\}$$

Posteriormente, vamos a aportar una definición alternativa.



# Capítulo 6

## Modelos de Herbrand alternativo

En este capítulo se da una definición alternativa del universo de Herbrand que simplifica su representación. El contenido de este capítulo se encuentra en el módulo `HerbrandAlt`

Se ha definido la alternativa en un capítulo aparte para luego decidir si mantenemos ambos capítulos complementándose o nos quedamos solo con este.

```
{-# LANGUAGE DeriveGeneric #-}
module Herbrand where
import Data.List
import Text.PrettyPrint.GenericPretty
import PFH
import LPH
import PTLP
```

Corregir codeEj para que lo considere doctest

### 6.1. Universo de Herbrand

**Definición 6.1.1.** Una **signatura** es una terna formada por las constantes, símbolos funcionales y símbolos de relación. Teniendo en cuenta la aridad tanto de los símbolos funcionales como los de relación.

Se define un tipo de dato para la signatura, cuya estructura es

$$( \text{constantes}, (\text{funciones}, \text{aridad}) , (\text{relaciones}, \text{aridad}) )$$

```
type Signatura = ([Nombre],[(Nombre,Int)],[(Nombre,Int)])
```

Dada una signatura, se procede a definir la función `(unionSignatura s1 s2)` que une las signaturas `s1` y `s2`.

```
-- | Ejemplos
-- >>> let s1 = ([ "a" ], [ ("f",1) ], [])
-- >>> let s2 = ([ "b", "c" ], [ ("f",1), ("g",2) ], [ ("R",2) ])
-- >>> unionSignatura s1 s2
-- ([ "a", "b", "c" ], [ ("f",1), ("g",2) ], [ ("R",2) ])
```

Su definición es

```
unionSignatura :: Signatura -> Signatura -> Signatura
unionSignatura (cs1,fs1,rs1) (cs2,fs2,rs2) =
    ( cs1 'union' cs2
    , fs1 'union' fs2
    , rs1 'union' rs2 )
```

Generalizamos la función anterior a la unión de una lista de signaturas mediante la función (`unionSignaturas ss`).

```
-- | Ejemplos
-- >>> let s1 = ([ "a" ], [ ("f",1) ], [])
-- >>> let s2 = ([ "b", "c" ], [ ("f",1), ("g",2) ], [ ("R",2) ])
-- >>> let s3 = ([ "a", "c" ], [], [ ("P",1) ])
-- >>> unionSignaturas [s1,s2,s3]
-- ([ "a", "b", "c" ], [ ("f",1), ("g",2) ], [ ("R",2), ("P",1) ])
```

Su definición es

```
unionSignaturas :: [Signatura] -> Signatura
unionSignaturas = foldr unionSignatura ([], [], [])
```

Se define la función (`signaturaTerm t`) que determina la signatura del término `t`.

```
-- | Ejemplos
-- >>> signaturaTerm tx
-- ([], [], [])
-- >>> signaturaTerm a
-- ([ "a" ], [], [])
-- >>> let t1 = Ter "f" [a,tx, Ter "g" [b,a]]
-- >>> t1
-- f[a,x,g[b,a]]
-- >>> signaturaTerm t1
-- ([ "a", "b" ], [ ("f",3), ("g",2) ], [])
```

Su definición es

```

signaturaTerm :: Termino -> Signatura
signaturaTerm (Var _) = ([], [], [])
signaturaTerm (Ter c []) = ([c], [], [])
signaturaTerm (Ter f ts) = (cs,[(f,n)] 'union' fs, rs)
    where
        (cs,fs,rs) = unionSignaturas (map signaturaTerm ts)
        n           = length ts

```

Una vez que podemos calcular la signatura de términos de una fórmula, se define la signatura de una fórmula  $f$  mediante la función (`signaturaForm f`).

```

-- | Ejemplos
-- >>> let f1 = Atom "R" [a,tx, Ter "g" [b,a]]
-- >>> f1
-- R[a,x,g[b,a]]
-- >>> signaturaForm f1
-- ([ "a", "b" ], [ ("g",2) ], [ ("R",3) ])
-- >>> signaturaForm (Neg f1)
-- ([ "a", "b" ], [ ("g",2) ], [ ("R",3) ])
-- >>> let f2 = Atom "P" [b]
-- >>> let f3 = Impl f1 f2
-- >>> f3
-- (R[a,x,g[b,a]]  $\implies$  P[b])
-- >>> signaturaForm f3
-- ([ "a", "b" ], [ ("g",2) ], [ ("R",3), ("P",1) ])
-- >>> let f4 = Conj [f1,f2,f3]
-- >>> f4
-- (R[a,x,g[b,a]]  $\wedge$  (P[b]  $\wedge$  (R[a,x,g[b,a]]  $\implies$  P[b])))
-- >>> signaturaForm f4
-- ([ "a", "b" ], [ ("g",2) ], [ ("R",3), ("P",1) ])
-- >>> let f5 = PTodo x f4
-- >>> f5
--  $\forall x$  (R[a,x,g[b,a]]  $\wedge$  (P[b]  $\wedge$  (R[a,x,g[b,a]]  $\implies$  P[b])))
-- >>> signaturaForm f5
-- ([ "a", "b" ], [ ("g",2) ], [ ("R",3), ("P",1) ])

```

Su definición es

```

signaturaForm :: Form -> Signatura
signaturaForm (Atom r ts) =
    (cs,fs, [(r,n)] 'union' rs)
    where (cs,fs,rs) = unionSignaturas (map signaturaTerm ts)
          n           = length ts
signaturaForm (Neg f) =
    signaturaForm f
signaturaForm (Impl f1 f2) =

```

```

    signaturaForm f1 'unionSignatura' signaturaForm f2
signaturaForm (Equiv f1 f2) =
    signaturaForm f1 'unionSignatura' signaturaForm f2
signaturaForm (Conj fs) =
    unionSignaturas (map signaturaForm fs)
signaturaForm (Disy fs) =
    unionSignaturas (map signaturaForm fs)
signaturaForm (PTodo _ f) =
    signaturaForm f
signaturaForm (Ex _ f) =
    signaturaForm f

```

Generalizamos la función anterior a una lista de fórmulas con la función (`signaturaForms fs`).

```

-- | Ejemplos
-- >>> let f1 = Atom "R" [Ter "f" [tx]]
-- >>> let f2 = Impl f1 (Atom "Q" [a, Ter "f" [b]])
-- >>> let f3 = Atom "S" [Ter "g" [a,b]]
-- >>> signaturaForms [f1,f2,f3]
-- ([ "a", "b" ], [ ("f",2), ("g",2) ], [ ("R",1), ("Q",2), ("S",1) ])

```

Su definición es

```

signaturaForms :: [Form] -> Signatura
signaturaForms fs =
    unionSignaturas (map signaturaForm fs)

```

El cálculo de la signatura de fórmulas y listas de ellas nos permite definir posteriormente el cálculo del universo de Herbrand.

**Definición 6.1.2.** El **universo de Herbrand** de  $L$  es el conjunto de términos básicos de  $F$ . Se representa por  $\mathcal{UH}(L)$ .

*Proposición 6.1.1.*  $\mathcal{UH}(L) = \bigcup_{i \geq 0} H_i(L)$ , donde  $H_i(L)$  es el nivel  $i$  del  $\mathcal{UH}(L)$  definido por

$$H_0(L) = \begin{cases} C, & \text{si } C \neq \emptyset \\ a, & \text{en caso contrario (a nueva constante)} \end{cases}$$

$$H_{i+1}(L) = H_i(L) \cup \{f(t_1, \dots, t_n) : f \in \mathcal{F}_n \text{ y } t_i \in H_i(L)\}$$

Definimos la función (`universoHerbrand n s`) que es el universo de Herbrand de la signatura  $s$  a nivel  $n$ .

```

-- | Ejemplos
-- >>> let s1 = ([ "a", "b", "c" ], [], [])

```

```

-- >>> universoHerbrand 0 s1
-- [a,b,c]
-- >>> universoHerbrand 1 s1
-- [a,b,c]
-- >>> let s2 = ([],[(f,1)],[])
-- >>> universoHerbrand 0 s2
-- [a]
-- >>> universoHerbrand 1 s2
-- [a,f[a]]
-- >>> universoHerbrand 2 s2
-- [a,f[a],f[f[a]]]
-- >>> let s3 = (["a","b"],[(f,1),(g,1)],[])
-- >>> universoHerbrand 0 s3
-- [a,b]
-- >>> universoHerbrand 1 s3
-- [a,b,f[a],f[b],g[a],g[b]]
-- >>> pp $ universoHerbrand 2 s3
-- [a,b,f[a],f[b],g[a],g[b],f[f[a]],f[f[b]],f[g[a]],
--   f[g[b]],g[f[a]],g[f[b]],g[g[a]],g[g[b]]]
-- >>> universoHerbrand 3 (["a"],[(f,1)],[(R,1)])
-- [a,f[a],f[f[a]],f[f[f[a]]]]
-- >>> pp $ universoHerbrand 3 (["a","b"],[(f,1),(g,1)],[])
-- [a,b,f[a],f[b],g[a],g[b],f[f[a]],f[f[b]],f[g[a]],
--   f[g[b]],g[f[a]],g[f[b]],g[g[a]],g[g[b]],f[f[f[a]]],
--   f[f[f[b]]],f[f[g[a]]],f[f[g[b]]],f[g[f[a]]],
--   f[g[f[b]]],f[g[g[a]]],f[g[g[b]]],g[f[f[a]]],
--   g[f[f[b]]],g[f[g[a]]],g[f[g[b]]],g[g[f[a]]],
--   g[g[f[b]]],g[g[g[a]]],g[g[g[b]]]]
-- >>> let s4 = (["a","b"],[(f,2)],[])
-- >>> universoHerbrand 0 s4
-- [a,b]
-- >>> universoHerbrand 1 s4
-- [a,b,f[a,a],f[a,b],f[b,a],f[b,b]]
-- >>> pp $ universoHerbrand 2 s4
-- [a,b,f[a,a],f[a,b],f[b,a],f[b,b],f[a,f[a,a]],
--   f[a,f[a,b]],f[a,f[b,a]],f[a,f[b,b]],f[b,f[a,a]],
--   f[b,f[a,b]],f[b,f[b,a]],f[b,f[b,b]],f[f[a,a],a],
--   f[f[a,a],b],f[f[a,a],f[a,a]],f[f[a,a],f[a,b]],
--   f[f[a,a],f[b,a]],f[f[a,a],f[b,b]],f[f[a,b],a],
--   f[f[a,b],b],f[f[a,b],f[a,a]],f[f[a,b],f[a,b]],
--   f[f[a,b],f[b,a]],f[f[a,b],f[b,b]],f[f[b,a],a],
--   f[f[b,a],b],f[f[b,a],f[a,a]],f[f[b,a],f[a,b]],
--   f[f[b,a],f[b,a]],f[f[b,a],f[b,b]],f[f[b,b],a],
--   f[f[b,b],b],f[f[b,b],f[a,a]],f[f[b,b],f[a,b]],
--   f[f[b,b],f[b,a]],f[f[b,b],f[b,b]]]

```

Su implementación es

```

universoHerbrand :: Int -> Signatura -> [Termino]
universoHerbrand 0 (cs,_,_)
  | null cs    = [a]
  | otherwise = [Ter c [] | c <- cs]
universoHerbrand n s@(_,fs,_) =
  u 'union'
  [Ter f ts | (f,k) <- fs
              , ts <- variacionesR k u]
  where u = universoHerbrand (n-1) s

```

Se define el universo de Herbrand de una fórmula  $f$  a nivel  $n$  mediante la función `(universoHerbrandForm n f)`.

```

-- | Ejemplos
-- >>> let f1 = Atom "R" [a,b,c]
-- >>> universoHerbrandForm 1 f1
-- [a,b,c]
-- >>> let f2 = Atom "R" [Ter "f" [tx]]
-- >>> universoHerbrandForm 2 f2
-- [a,f[a],f[f[a]]]
-- >>> let f3 = Impl f2 (Atom "Q" [a,Ter "g" [b]])
-- >>> f3
-- (R[f[x]] $\implies$ Q[a,g[b]])
-- >>> pp $ universoHerbrandForm 2 f3
-- [a,b,f[a],f[b],g[a],g[b],f[f[a]],f[f[b]],f[g[a]],
--   f[g[b]],g[f[a]],g[f[b]],g[g[a]],g[g[b]]]
-- >>> let f4 = Atom "R" [Ter "f" [a,b]]
-- >>> signaturaForm f4
-- ([a,b],[(f,2)],[(R,1)])
-- >>> pp $ universoHerbrandForm 2 f4
-- [a,b,f[a,a],f[a,b],f[b,a],f[b,b],f[a,f[a,a]],
--   f[a,f[a,b]],f[a,f[b,a]],f[a,f[b,b]],f[b,f[a,a]],
--   f[b,f[a,b]],f[b,f[b,a]],f[b,f[b,b]],f[f[a,a],a],
--   f[f[a,a],b],f[f[a,a],f[a,a]],f[f[a,a],f[a,b]],
--   f[f[a,a],f[b,a]],f[f[a,a],f[b,b]],f[f[a,b],a],
--   f[f[a,b],b],f[f[a,b],f[a,a]],f[f[a,b],f[a,b]],
--   f[f[a,b],f[b,a]],f[f[a,b],f[b,b]],f[f[b,a],a],
--   f[f[b,a],b],f[f[b,a],f[a,a]],f[f[b,a],f[a,b]],
--   f[f[b,a],f[b,a]],f[f[b,a],f[b,b]],f[f[b,b],a],
--   f[f[b,b],b],f[f[b,b],f[a,a]],f[f[b,b],f[a,b]],
--   f[f[b,b],f[b,a]],f[f[b,b],f[b,b]]]

```

A continuación, su definición es

```

universoHerbrandForm :: Int -> Form -> [Termino]
universoHerbrandForm n f =
  universoHerbrand n (signaturaForm f)

```



Se generaliza la definición anterior a una lista de fórmulas mediante la función `(universoHerbrandForms n fs)`

```
-- | Ejemplos
-- >>> let f1 = Atom "R" [Ter "f" [tx]]
-- >>> let f2 = Impl f1 (Atom "Q" [a, Ter "f" [b]])
-- >>> let f3 = Atom "S" [Ter "g" [a, b]]
-- >>> universoHerbrandForms 1 [f1, f2, f3]
-- [a, f[a], b, f[b], g[a, a], g[a, b], g[b, a], g[b, b]]
```

Siendo su definición

```
universoHerbrandForms :: Int -> [Form] -> [Termino]
universoHerbrandForms n fs =
  nub (concatMap (universoHerbrandForm n) fs)
```

*Proposición 6.1.2.*  $\mathcal{UH}$  es finito si y sólo si no tiene símbolos de función.

## 6.2. Base de Herbrand

**Definición 6.2.1.** Una **fórmula básica** es una fórmula sin variables ni cuantificadores.

**Definición 6.2.2.** La **base de Herbrand**  $\mathcal{BH}(L)$  de un lenguaje  $L$  es el conjunto de átomos básicos de  $L$ .

Definimos un tipo de dato para las bases de Herbrand

```
type BaseH = [Form]
```

Implementamos la base de herbrand a nivel  $n$  de la signatura  $s$  mediante la función `(baseHerbrand n s)`

```
-- | Ejemplos
-- >>> let s1 = (["a", "b", "c"], [], [("P", 1)])
-- >>> baseHerbrand 0 s1
-- [P[a], P[b], P[c]]
-- >>> let s2 = (["a", "b", "c"], [], [("P", 1), ("Q", 1), ("R", 1)])
-- >>> let s2 = (["a", "b", "c"], [("f", 1)], [("P", 1), ("Q", 1), ("R", 1)])
-- >>> baseHerbrand 0 s2
-- [P[a], P[b], P[c], Q[a], Q[b], Q[c], R[a], R[b], R[c]]
-- >>> pp $ baseHerbrand 1 s2
-- [P[a], P[b], P[c], P[f[a]], P[f[b]], P[f[c]], Q[a], Q[b],
--   Q[c], Q[f[a]], Q[f[b]], Q[f[c]], R[a], R[b], R[c], R[f[a]],
--   R[f[b]], R[f[c]]]
```

Se implementa en Haskell a continuación

```
baseHerbrand :: Int -> Signatura -> BaseH
baseHerbrand n s@(_,_,rs) =
  [Atom r ts | (r,k) <- rs
               , ts <- variacionesR k u]
  where u = universoHerbrand n s
```

Se define la base de Herbrand de una fórmula  $f$  a nivel  $n$  mediante (`baseHerbrandForm n f`).

```
-- | Ejemplos
-- >>> let f1 = Atom "P" [Ter "f" [tx]]
-- >>> f1
-- P[f[x]]
-- >>> baseHerbrandForm 2 f1
-- [P[a],P[f[a]],P[f[f[a]]]]
```

Su definición es

```
baseHerbrandForm :: Int -> Form -> BaseH
baseHerbrandForm n f =
  baseHerbrand n (signaturaForm f)
```

Generalizamos la función anterior a una lista de fórmulas definiendo (`baseHerbrandForms n fs`)

```
-- | Ejemplos
-- >>> let f1 = Atom "P" [Ter "f" [tx]]
-- >>> let f2 = Atom "Q" [Ter "g" [b]]
-- >>> baseHerbrandForms 1 [f1,f2]
-- [P[b],P[f[b]],P[g[b]],Q[b],Q[f[b]],Q[g[b]]]
```

Su definición es

```
baseHerbrandForms :: Int -> [Form] -> BaseH
baseHerbrandForms n fs =
  baseHerbrandForm n (Conj fs)
```

## 6.3. Interpretacion de Herbrand

**Definición 6.3.1.** Una interpretación de Herbrand es una interpretación  $\mathcal{I} = (\mathcal{U}, I)$  tal que

- $\mathcal{U}$  es el universo de Herbrand de  $F$ .
- $I(c) = c$ , para constante  $c$  de  $F$ .
- $I(f) = f$ , para cada símbolo funcional de  $F$ .

Definimos un tipo de dato para los elementos que componen la interpretación de Herbrand.

```
type UniversoH = Universo Termino

type InterpretacionHR = InterpretacionR Termino

type InterpretacionHF = InterpretacionF Termino

type InterpretacionH = (InterpretacionHR, InterpretacionHF)
```

```
type AtomosH = [Form]
```

### Definir fórmulas atómicas básicas

Se define la interpretación de Herbrand de un conjunto de átomos de Herbrand a través de `(interpretacionH fs)`

```
-- | Ejemplos
-- >>> let f1 = Atom "P" [a]
-- >>> let f2 = Atom "P" [Ter "f" [a,b]]
-- >>> let fs = [f1,f2]
-- >>> let (iR,iF) = interpretacionH fs
-- >>> iF "f" [a,c]
-- f[a,c]
-- >>> iR "P" [a]
-- True
-- >>> iR "P" [b]
-- False
-- >>> iR "P" [Ter "f" [a,b]]
-- True
-- >>> iR "P" [Ter "f" [a,a]]
-- False
```

Se implementa en Haskell

```

interpretacionH :: AtomosH -> InterpretacionH
interpretacionH fs = (iR,iF)
  where iF f ts    = Ter  f ts
        iR r ts    = Atom r ts 'elem' fs

```

*Proposición 6.3.1.* Una interpretación de Herbrand queda determinada por un subconjunto de la base de Herbrand.

Evaluamos una fórmula a través de una interpretación de Herbrand. Para ello definimos la función  $(\text{valorH } u \ i \ f)$ ; donde  $u$  representa el universo,  $i$  la interpretación, y  $f$  la fórmula.

```

valorH :: UniversoH -> InterpretacionH -> Form -> Bool
valorH u i f =
  valorF u i s f
  where s _ = a

```

## 6.4. Modelos de Herbrand

**Definición 6.4.1.** Un **modelo de Herbrand** de una fórmula  $F$  es una interpretación de Herbrand de  $F$  que es modelo de  $F$ .

Un **modelo de Herbrand** de un conjunto de fórmulas  $S$  es una interpretación de Herbrand de  $S$  que es modelo de  $S$ .

Implementamos los subconjuntos del  $n$ -ésimo nivel de la base de Herbrand de la fórmula  $f$  que son modelos de  $f$  con la función  $(\text{modelosHForm } n \ f)$ .

```

-- | Ejemplos
-- >>> let f1 = Disy [Atom "P" [a], Atom "P" [b]]
-- >>> f1
-- (P[a] ∨ P[b])
-- >>> modelosHForm 0 f1
-- [[P[a]], [P[b]], [P[a], P[b]]]
-- >>> let f2 = Impl (Atom "P" [a]) (Atom "P" [b])
-- >>> f2
-- (P[a] ⇒ P[b])
-- >>> modelosHForm 0 f2
-- [[], [P[b]], [P[a], P[b]]]
-- >>> let f3 = Conj [Atom "P" [a], Atom "P" [b]]
-- >>> f3
-- (P[a] ∧ P[b])
-- >>> modelosHForm 0 f3

```

```
-- [[P[a],P[b]]]
-- >>> let f4 = PTodo x (Impl (Atom "P" [tx]) (Atom "Q" [Ter "f" [tx]]))
-- >>> f4
--  $\forall x (P[x] \implies Q[f[x]])$ 
-- >>> modelosHForm 0 f4
-- [[],[Q[a]]]
-- >>> modelosHForm 1 f4
-- [[],[Q[a]], [Q[f[a]]], [P[a],Q[f[a]]], [Q[a],Q[f[a]]], [P[a],Q[a],Q[f[a]]]]
-- >>> length (modelosHForm 2 f4)
-- 18
```

Lo definimos en Haskell

```
modelosHForm :: Int -> Form -> [AtomosH]
modelosHForm n f =
  [fs | fs <- subsequences bH
    , valorH uH (interpretacionH fs) f]
  where uH = universoHerbrandForm n f
        bH = baseHerbrandForm n f
```

Generalizamos la definición anterior a una lista de fórmulas mediante la función (modelosH n fs).

```
-- | Ejemplos
-- >>> let f1 = PTodo x (Impl (Atom "P" [tx]) (Atom "Q" [Ter "f" [tx]]))
-- >>> f1
--  $\forall x (P[x] \implies Q[f[x]])$ 
-- >>> let f2 = Ex x (Atom "P" [tx])
-- >>> f2
--  $\exists x P[x]$ 
-- >>> modelosH 0 [f1,f2]
-- []
-- >>> modelosH 1 [f1,f2]
-- [[P[a],Q[f[a]]], [P[a],Q[a],Q[f[a]]]]
```

Su definición es

```
modelosH :: Int -> [Form] -> [AtomosH]
modelosH n fs =
  [gs | gs <- subsequences bH
    , and [valorH uH (interpretacionH gs) f | f <- fs]]
  where uH = universoHerbrandForms n fs
        bH = baseHerbrandForms n fs
```



# Capítulo 7

## Resolución en lógica de primer orden

En este capítulo se introducirá la resolución en la lógica de primer orden y se implementará en Haskell. El contenido de este capítulo se encuentra en el módulo RES

```
module RES where
import Data.List
import LPH
import DNH
import PTLP
```

### 7.1. Ampliación de la forma clausal

La resolución requiere del uso de las formas clausales definidas anteriormente. Definamos algunas cuestiones:

Primero implementemos un tipo de dato adecuado para las interpretaciones de cláusulas, InterpretacionC.

```
type InterpretacionC = [(Form,Int)]
```

**Definición 7.1.1.** El **valor** de una cláusula  $C$  en una interpretación  $I$  es

$$I(C) = \begin{cases} 1, & \text{si existe un } L \in C \text{ tal que } I(L) = 1, \\ 0, & \text{en caso contrario} \end{cases}$$

Implementamos el valor de una cláusula  $C$  mediante la función (`valorC c is`)

```
valorC :: Clausula -> InterpretacionC -> Int
valorC (C fs) is =
  if ([1 | (f,1) <- is, elem f fs] ++
      [1 | (f,0) <- is, elem (Neg f) fs]) /= []
  then 1 else 0
```

**Definición 7.1.2.** El **valor** de un conjunto de cláusulas  $S$  en una interpretación  $I$  es

$$I(S) = \begin{cases} 1, & \text{si para toda } C \in S, I(C) = 1, \\ 0, & \text{en caso contrario} \end{cases}$$

Implementamos el valor de un conjunto de cláusulas mediante la función (`valorCs` `s is`)

```
valorCs :: Clausulas -> InterpretacionC -> Int
valorCs (Cs cs) is =
    if (all (==1) [valorC c is | c <- cs]) then 1 else 0
```

*Nota 7.1.1.* Cualquier interpretación de la cláusula vacía es 0.

**Definición 7.1.3.** Una cláusula  $C$  y una fórmula  $F$  son **equivalentes** si  $I(C) = I(F)$  para cualquier interpretación  $I$ .

Veamos algunos ejemplos que nos ilustren lo definido hasta ahora:

```
-- | Ejemplos
-- >>> let c = Cs [C [Neg p,p],C [Neg p,Neg r]]
-- >>> c
-- {{¬p,p},{¬p,¬r}}
-- >>> valorCs c [(p,1),(r,0)]
-- 1
-- >>> valorCs c [(p,1),(r,1)]
-- 0
```

**Definición 7.1.4.** Una interpretación  $I$  es **modelo** de un conjunto de cláusulas  $S$  si  $I(S) = 1$ .

Caracterizamos el concepto de modelo de un conjunto de cláusulas mediante la función (`is 'esModeloDe' cs`)

```
esModeloDe :: InterpretacionC -> Clausulas -> Bool
esModeloDe is cs = valorCs cs is == 1
```

```
-- | Ejemplos
-- >>> let c = Cs [C [Neg p,p],C [Neg p,Neg r]]
-- >>> let is = [(p,1),(r,0)]
-- >>> is 'esModeloDe' c
-- True
```



**Definición 7.1.5.** Un conjunto de cláusulas es **consistente** si tiene modelos e **inconsistente**, en caso contrario.

Definamos una serie de funciones necesarias para determinar si un conjunto de cláusulas es consistente.

Primero definimos la función `(atomosC c)` y `(atomosCs cs)` que obtienen una lista de los átomos que aparecen en la cláusula o conjuntos de cláusulas `c` y `cs`, respectivamente.

```
atomosC :: Clausula -> [Form]
atomosC (C fs) = nub ([f | f <- fs, esAtomo f] ++ [f | (Neg f) <- fs])
  where
    esAtomo (Atom _ _) = True
    esAtomo _ = False

atomosCs :: Clausulas -> [Form]
atomosCs (Cs cs) = nub (concat [atomosC c | c <- cs])
```

A continuación, mediante la implementación de `(interPosibles cs)` obtenemos una lista de todas las posibles interpretaciones que podemos obtener de los átomos de `cs`.

```
interPosibles :: Clausulas -> [InterpretacionC]
interPosibles = sequence . aux2 . aux1 . atomosCs
  where
    aux1 fs = [(a,b) | a <- fs, b <- [0,1]]
    aux2 [] = []
    aux2 fs = [take 2 fs] ++ (aux2 (drop 2 fs))
```

Finalmente, comprobamos con la función `(esConsistente cs)` que alguna de las interpretaciones posibles es modelo del conjunto de cláusulas.

```
esConsistente :: Clausulas -> Bool
esConsistente cs = or [i 'esModeloDe' cs | i <- is]
  where
    is = interPosibles cs
```

Ahora, como acostumbramos, veamos algunos ejemplos de las funciones definidas.

```
-- | Ejemplos
-- >>> let c = Cs [C [Neg p,p],C [Neg p,Neg r]]
-- >>> atomosCs c
-- [p,r]
-- >>> interPosibles c
-- [(p,0),(r,0)],[(p,0),(r,1)],[(p,1),(r,0)],[(p,1),(r,1)]
-- >>> esConsistente c
-- True
```



## Capítulo 8

# Correspondencia de Curry-Howard

En este capítulo trataremos la correspondencia de Curry-Howard, también llamada isomorfismo o equivalencia de Curry-Howard. Debe su nombre a que establece una correspondencia entre las pruebas en la lógica y los tipos de datos en la programación.

Comencemos un módulo en Haskell, donde escribiremos nuestros ejemplos.

```
module CHC where
import LPH
import Data.Void
import Data.Either
```

En Haskell, podemos definir una gran diversidad de funciones que manejan tipos distintos, por ejemplo:

```
-- >>> :t map
-- map :: (a -> b) -> [a] -> [b]
-- >>> :t curry
-- curry :: ((a, b) -> c) -> a -> b -> c
-- >>> :t elem
-- elem :: (Eq a, Foldable t) => a -> t a -> Bool
```

Por ello, cabe preguntarnos si existen funciones para todo tipo que nos podamos inventar. Ahí entra en juego la correspondencia de Curry-Howard, estableciendo las siguientes equivalencias:

- Los tipos son teoremas.
- Los programas son demostraciones.

Lo que indica que existirá una función de un determinado tipo si dicho tipo, interpretado como una proposición lógica, es cierto.

Mostremos en una tabla, aunque posteriormente tratemos con ejemplos y más detenimiento, las correspondencias entre elementos propios de la lógica y elementos del  $\lambda$ -cálculo, en concreto Haskell.

Lógica	Haskell
$\rightarrow$	<code>-&gt;</code>
$a \wedge b$	<code>(a,b)</code>
$a \vee b$	<code>Either a b</code>
$T$	<code>()</code>
$F$	<code>Void</code>
$\neg$	<code>not</code>
Modus ponens	Aplicación de funciones

Veamos en el siguiente ejemplo el isomorfismo de Curry-Howard, así como la interpretación que se infiere de ella. Por ejemplo, si tenemos la proposición lógica

$$a \rightarrow a$$

En Haskell, dicha proposición equivale al tipo de programa

$$f :: a \rightarrow a$$

Dicho tipo de dato se puede traducir como que si el tipo  $a$  “está habitado” entonces, como es evidente,  $a$  también lo está. Y la prueba de dicha proposición lógica equivale a la existencia de una función con dicho tipo, en este caso la función identidad (`id :: a -> a`). Otro ejemplo sencillo puede ser la función `show`, cuyo tipo de dato si preguntamos a Haskell es

```
-- >>> :t show
-- show :: Show a => a -> String
```

Que no es más que la proposición lógica  $(\forall a \in P. a \rightarrow b)$ . En este caso,  $a$  es cualquier tipo que tenga la propiedad  $P$ ,  $P$  es la clase de los datos que se pueden mostrar y  $b$  es el tipo de dato `String`.

Pensemos ahora en dirección contraria. Si tenemos la función composición `(.)` y preguntamos a Haskell por su tipo de dato obtenemos lo siguiente:

```
-- >>> :t (.)
-- (.) :: (b -> c) -> (a -> b) -> a -> c
```

Que con una ligera reflexión, no es más que la proposición lógica que afirma que

$$\forall a, b, c \in P. (b \rightarrow c) \rightarrow (a \rightarrow b) \rightarrow (a \rightarrow c)$$

Saquemos una serie de conclusiones de los ejemplos y, posteriormente, tratemos las distintas entradas de la tabla anterior en secciones separadas.

- Dada una proposición lógica podemos inferir un tipo de dato equivalente a ella.
- Demostrar un teorema es equivalente a encontrar una función del tipo de dato adecuado.

Referencia a clases en Haskell



## **Parte II**

### **Sistemas utilizados**





# Capítulo 9

## Apéndice: GitHub

En este apéndice se pretende introducir al lector en el empleo de [GitHub](#), sistema remoto de versiones.

### 9.1. Crear una cuenta

El primer paso es crear una cuenta en la página web de [GitHub](#), para ello `sign up` y se rellena el formulario.

### 9.2. Crear un repositorio

Mediante `New repository` se crea un repositorio nuevo. Un repositorio es una carpeta de trabajo. En ella se guardarán todas las versiones y modificaciones de nuestros archivos.

Necesitamos darle un nombre adecuado y seleccionar

1. En `Add .gitignore` se selecciona Haskell.
2. En `add a license` se selecciona GNU General Public License v3.0.

Finalmente `Create repository`

### 9.3. Conexión

Nuestro interés está en poder trabajar de manera local y poder tanto actualizar GitHub como nuestra carpeta local. Los pasos a seguir son

1. Generar una clave SSH mediante el comando

```
ssh-keygen -t rsa -b 4096 -C "tuCorreo"
```

Indicando una contraseña. Si queremos podemos dar una localización de guardado de la clave pública.

2. Añadir la clave a `ssh-agent`, mediante

```
eval "$(ssh-agent -s)"
ssh-add ~/.ssh/id_rsa
```

3. Añadir la clave a nuestra cuenta. Para ello: Setting → SSH and GPG keys → New SSH key. En esta última sección se copia el contenido de

```
~/.ssh/id_rsa.pub
```

por defecto. Podríamos haber puesto otra localización en el primer paso.

4. Se puede comprobar la conexión mediante el comando

```
ssh -T git@github.com
```

5. Se introducen tu nombre y correo

```
git config --global user.name "Nombre"
git config --global user.email "<tuCorreo>"
```

## 9.4. Pull y push

Una vez hecho los pasos anteriores, ya estamos conectados con GitHub y podemos actualizar nuestros ficheros. Nos resta aprender a actualizarlos.

1. Clonar el repositorio a una carpeta local:

Para ello se ejecuta en una terminal

```
git clone <enlace que obtienes en el repositorio>
```

El enlace que sale en el repositorio pinchando en (clone or download) y, eligiendo (use SSH).

2. Actualizar tus ficheros con la versión de GitHub:

En emacs se ejecuta (Esc-x)+(magit-status). Para ver una lista de los cambios que están (unpulled), se ejecuta en magit remote update. Se emplea pull, y se actualiza. (Pull: origin/master)

3. Actualizar GitHub con tus archivos locales:

En emacs se ejecuta (Esc-x)+(magit-status). Sale la lista de los cambios (UnStages). Con la (s) se guarda, la (c)+(c) hace (commit). Le ponemos un título y, finalmente (Tab+P)+(P) para hacer (push) y subirlos a GitHub.

## 9.5. Ramificaciones (“branches”)

Uno de los puntos fuertes de Git es el uso de ramas. Para ello, creamos una nueva rama de trabajo. En (`magit-status`), se pulsa `b`, y luego `(c)` (`create a new branch and checkout`). Checkout cambia de rama a la nueva, a la que habrá que dar un nombre.

Se trabaja con normalidad y se guardan las modificaciones con `magit-status`. Una vez acabado el trabajo, se hace (`merge`) con la rama principal y la nueva.

Se cambia de rama (`branch...`) y se hace (`pull`) como acostumbramos.

Finalmente, eliminamos la rama mediante (`magit-status`)  $\rightarrow$  (`b`)  $\rightarrow$  (`k`)  $\rightarrow$  (Nombre de la nueva rama)

# Índice alfabético

alfa, 69  
algun, 15  
apPred, 85  
aplicaFunAConst, 81  
aplicaFun, 81  
aplicaPred, 85  
aplicafun, 14  
aquellosQuecumplen, 14  
aridadF, 81  
atomosCs, 105  
atomosC, 105  
baseHerbrandForms, 98  
baseHerbrandForm, 98  
baseHerbrand, 85, 97  
beta, 69  
componentes, 73  
composicion, 44  
conjuncion, 15  
constantesForm, 80  
constantesTerm, 80  
contieneLaLetra, 12  
contradiccion, 47  
cuadrado, 11  
delta, 70  
descomponer, 74  
disyuncion, 15  
divideEntre2, 18  
divisiblePor, 12  
dobleNeg, 73  
dominio, 42  
elemMap, 48  
elimImpEquiv, 57  
eliminaCuant, 62  
enFormaNC, 56  
esCerrado, 77  
esConsistente, 105  
esConstante, 79  
esFuncion, 81  
esModeloDe, 104  
esNodoCerrado, 76  
esPredicado, 84  
esTeorema, 77  
esVariable, 30  
expandeTableroG, 76  
expandeTablero, 76  
factorial, 16  
form3CAC, 68  
formaNPConjuntiva, 63  
formaNormalPrenexa, 63  
formaRectificada, 61  
formulaAbierta, 40  
funForm, 81  
gamma, 69  
hacerApropiada, 42  
identidad, 41  
interPosibles, 105  
interiorizaNeg, 58  
interpretacionH, 99  
listaInversa, 17  
literalATer, 73  
literal, 70  
modelosHForm, 100  
noPertenece, 16  
nodoExpandido, 75

pertenece, 16  
plegadoPorlaDerecha, 16  
plegadoPorlaIzquierda, 17  
predicadosForm, 84  
primerElemento, 14  
quita, 48  
raizCuadrada, 12  
ramificacionP, 75  
ramificacion, 74  
recolectaCuant, 62  
recorrido, 42  
reflexiva, 26  
refuta, 77  
satisfacible, 77  
segundoElemento, 14  
signaturaForms, 94  
signaturaForm, 93  
signaturaTerm, 92  
simbolosPred, 85  
simetrica, 26  
skfs, 66  
skf, 65  
skolem, 66  
skol, 65  
sk, 66  
sumaLaLista, 16  
susTab, 76  
susTerms, 43  
susTerm, 43  
sustAux, 61  
sustNodo, 76  
sustitucionForm, 44  
sustituyeVar, 42  
sustituye, 28  
tableroInicial, 77  
todos, 15  
unificaConjuncion, 59  
unionSignaturas, 92  
unionSignatura, 91  
univHerbrand, 82  
universoHerbrandForms, 97  
universoHerbrandForm, 96  
universoHerbrand, 94  
valHerbrand, 87  
valorCs, 104  
valorC, 103  
valorF, 34  
valorHerbrand, 87  
valorH, 100  
valorT, 32  
valor, 28  
varEnForm, 39  
varEnTerms, 38  
varEnTerm, 38  
varLigada, 74  
variacionesR, 13



# Lista de tareas pendientes

■ sección en proceso . . . . .	20
■ Pendiente de revisión. . . . .	35
■ Sería interesante comparar la representación de sustituciones mediante diccionarios con la librería Data.Map . . . . .	41
■ No compatible con nuestra definición de término, hay que adaptarlo . . . . .	46
■ Distinguir el caso de fórmulas con variables libres. . . . .	69
■ Se solapan las ramas del arbol . . . . .	71
■ Explicar más el método de tableros con polaridad. . . . .	78
■ Comparar la implementación con la de Ben Ari que se encuentra en <a href="https://github.com/motib/mlcs/blob/master/fol/tab1.pro">https://github.com/motib/mlcs/blob/master/fol/tab1.pro</a> . . . . .	78
■ Definir directamente simbolosPred sin usar predForm y eliminar predForm y predicadosForm. Respuesta a comentario: La razón de no haber definido simbolosPred sin predForm y predicadosForm es por la necesidad de mantener la aridad del predicado para la definición de base de Herbrand. . . . .	85
■ Se ha definido la alternativa en un capítulo aparte para luego decidir si mantenemos ambos capítulos complementándose o nos quedamos solo con este. . . . .	91
■ Corregir codeEj para que lo considere doctest . . . . .	91
■ Definir fórmulas atómicas básicas . . . . .	99
■ Referencia a clases en Haskell . . . . .	109